code:
MAP_NAME = "夕暮港"
MAP_IMAGE = "duskport.webp"
PATH_COLOR = "0xf85858"
PATH_ALPHA = "0x7d"
NODE_BASE_COLOR = "0x454ff"

repo_name: napo/erafl

license: unknown


code:
<DocumentElement>
  <GMAPDATA>
    <id>638179119713683404</id>
    <NODE_ID>14</NODE_ID>
    <NODE_NAME>未使用</NODE_NAME>
    <POS_X>540</POS_X>
    <POS_Y>220</POS_Y>
    <PATH_LIST />
  </GMAPDATA>
  <GMAPDATA>
    <id>638179120173496576</id>
    <NODE_ID>13</NODE_ID>
    <NODE_NAME>未使用</NODE_NAME>
    <POS_X>540</POS_X>
    <POS_Y>200</POS_Y>
    <PATH_LIST />
  </GMAPDATA>
  <GMAPDATA>
    <id>638179120516476280</id>
    <NODE_ID>12</NODE_ID>
    <NODE_NAME>新市街</NODE_NAME>
    <POS_X>400</POS_X>
    <POS_Y>340</POS_Y>
    <PATH_LIST>9,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638179121013464574</id>
    <NODE_ID>11</NODE_ID>
    <NODE_NAME>政庁区</NODE_NAME>
    <POS_X>380</POS_X>
    <POS_Y>120</POS_Y>
    <PATH_LIST>4,7,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638179121148667968</id>
    <NODE_ID>10</NODE_ID>
    <NODE_NAME>漁業区</NODE_NAME>
    <POS_X>220</POS_X>
    <POS_Y>300</POS_Y>
    <PATH_LIST>7,3,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638179121279666951</id>
    <NODE_ID>15</NODE_ID>
    <NODE_NAME>未使用</NODE_NAME>
    <POS_X>540</POS_X>
    <POS_Y>240</POS_Y>
    <PATH_LIST />
  </GMAPDATA>
  <GMAPDATA>
    <id>638179122091232299</id>
    <NODE_ID>17</NODE_ID>
    <NODE_NAME>未使用</NODE_NAME>
    <POS_X>540</POS_X>
    <POS_Y>280</POS_Y>
    <PATH_LIST />
  </GMAPDATA>
  <GMAPDATA>
    <id>638179122388305632</id>
    <NODE_ID>6</NODE_ID>
    <NODE_NAME>船着場</NODE_NAME>
    <POS_X>160</POS_X>
    <POS_Y>240</POS_Y>
    <PATH_LIST>5,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638179123574879005</id>
    <NODE_ID>7</NODE_ID>
    <NODE_NAME>潮騒広場地区</NODE_NAME>
    <POS_X>300</POS_X>
    <POS_Y>220</POS_Y>
    <PATH_LIST>2,3,10,4,11,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638179124042393830</id>
    <NODE_ID>8</NODE_ID>
    <NODE_NAME>鉄道駅</NODE_NAME>
    <POS_X>240</POS_X>
    <POS_Y>60</POS_Y>
    <PATH_LIST>0,1,2,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638179124383754194</id>
    <NODE_ID>2</NODE_ID>
    <NODE_NAME>通関局</NODE_NAME>
    <POS_X>220</POS_X>
    <POS_Y>180</POS_Y>
    <PATH_LIST>8,7,1,5,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638179414069835197</id>
    <NODE_ID>0</NODE_ID>
    <NODE_NAME>大陸鉄道</NODE_NAME>
    <POS_X>240</POS_X>
    <POS_Y>20</POS_Y>
    <PATH_LIST>8,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638179424942789360</id>
    <NODE_ID>1</NODE_ID>
    <NODE_NAME>フォアマスト塔地区</NODE_NAME>
    <POS_X>160</POS_X>
    <POS_Y>120</POS_Y>
    <PATH_LIST>8,2,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638179425567006028</id>
    <NODE_ID>3</NODE_ID>
    <NODE_NAME>街商通り</NODE_NAME>
    <POS_X>300</POS_X>
    <POS_Y>300</POS_Y>
    <PATH_LIST>7,10,9,4,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638179426017058788</id>
    <NODE_ID>4</NODE_ID>
    <NODE_NAME>ミズンマスト塔地区</NODE_NAME>
    <POS_X>400</POS_X>
    <POS_Y>260</POS_Y>
    <PATH_LIST>7,3,11,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638179426856470682</id>
    <NODE_ID>5</NODE_ID>
    <NODE_NAME>埠頭地区</NODE_NAME>
    <POS_X>180</POS_X>
    <POS_Y>200</POS_Y>
    <PATH_LIST>6,2,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638179427633298890</id>
    <NODE_ID>9</NODE_ID>
    <NODE_NAME>陸門</NODE_NAME>
    <POS_X>340</POS_X>
    <POS_Y>360</POS_Y>
    <PATH_LIST>3,12,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638179428247180521</id>
    <NODE_ID>16</NODE_ID>
    <NODE_NAME>未使用</NODE_NAME>
    <POS_X>540</POS_X>
    <POS_Y>260</POS_Y>
  </GMAPDATA>
</DocumentElement>


code:
;-------------------------------------------------
;クエスト定義ファイル
;QUEST_22:ノアレルム辺境の賊徒
;-------------------------------------------------
@QST_TITLE_QUEST_22
RESULTS '= "ノアレルム辺境の賊徒"

;ギルドメニューでクエストをフィルターする際のタグリスト。末尾の','を忘れないように
@QST_QUEST_TAG_22
RESULTS '= "周期,探索,"

;クエスト選択画面で呼び出される概要説明
;クエストメニュー右上段の概要欄。10行まで
@QST_SUMALLY1_QUEST_22
RESULTS:0 '= "ノアレルム共和国からの依頼だ"
RESULTS:1 '= "曰く、ノアレルム北部辺境地域に現れる賊徒が、集落を荒らして去っていく事件が繰り返されているとのこと"
RESULTS:2 '= "共和国は専らリナシメントに軍主力を投入しているため正規兵を割く余裕がない"
RESULTS:3 '= "そこで、報奨金をかけて冒険者に賊を狩り立てさせようとしているようだ"
RESULTS:5 '= "しかし、聞き及んだ限りでは既に数組の討手が出たにも関わらず、まだ事態は終息に向かっていない"
RESULTS:6 '= "どうも賊は相当の腕利きのようだ"
RESULTS:7 '= "賊の根城に迫るには、それなりのスキルを持った探索員が必要かもしれない……"

;クエストメニュー右下段の備考欄。難易度や条件、報酬等を記述。7行まで
@QST_SUMALLY2_QUEST_22
RESULTS:0 '= @"難易度：@C:{カラー_注意}@★×８@/C@"
RESULTS:1 '= "報酬：20,000ディナリ～"
RESULTS:2 '= "最低パーティ人数：2人　　　推奨パーティ人数:5人以上　　　推奨平均戦闘値：80以上"
RESULTS:3 '= "備考:定期性クエスト(4期毎)"
RESULTS:4 '= @"@C:{カラー_選択不可}@「……ニンジャ、って知ってるか？　奇怪な術と人間離れした身体能力で敵を皆殺しにするっていうぜ」@/C@"
RESULTS:5 '= @"@C:{カラー_選択不可}@「ま、いわゆる東方の神秘ってやつだ。それが、今回の件に絡んでるらしい。信じるか？」@/C@"
RESULTS:6 '= @"@C:{カラー_選択不可}@「初めは俺も笑ってたさ――ギルドの腕っこきの連中が返り討ちに合うまでは、な」@/C@"

;-------------------------------------------------
;クエストの実行可能条件を定義
;-------------------------------------------------
@QST_AVAILABLE_QUEST_22(QUEST_ID)
#DIM QUEST_ID
;リナシメントの進捗スコア記録6以上で解禁
SIF QST_SV_QUEST_FLAG:9 <= 5
	RETURN 0
;ススキが仲間入りしていると解禁
SIF CFLAG:NAME_TO_CHARA("UC_ススキ"):所属 != CFLAG:MASTER:所属
	RETURN 0
SIF QST_COOLTIME:QUEST_ID
	RETURN 0
RETURN 1

;-------------------------------------------------
;クエスト固有の開始条件
;[このクエストを開始する]押下時に呼ばれる
;ススキがパーティ入りしている必要あり
;-------------------------------------------------
@QST_CAN_START_QUEST_22(QUEST_ID)
#DIM QUEST_ID
#DIM DYNAMIC MEMBER
#DIM DYNAMIC CHARA_ID
#DIM DYNAMIC NUM
LOCALS '= ""

FOR LOCAL, 0, LIST_COUNT(_LIST_QST_PARTY)
	CHARA_ID = TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))
	IF CHARA_ID == NAME_TO_CHARA("UC_ススキ")
		MEMBER = 1
	ENDIF
	NUM ++
NEXT
IF MEMBER == 0
	CALLF TAG_PRINT(@"@C:{カラー_注意}@必要メンバーが選出されていません(必須編成メンバー：ススキ)@/C@", 1)
	WAIT
	RETURN 0
ELSEIF NUM < 2
	CALLF TAG_PRINT(@"@C:{カラー_注意}@パーティ人数が不足しています(必須：２名以上)@/C@", 1)
	WAIT
	RETURN 0
ELSE
	RETURN 1
ENDIF

;-------------------------------------------------
;マップデータ定義
;クエストタイプが"GMAPの場合"マップ定義は外部マップデータに持っているので
; ERB側のマップデータにはEVENT_LISTさえ定義すればよい
;-------------------------------------------------
@QST_SET_MAPDATA_QUEST_22(MAP_ID)
#DIM MAP_ID
;mapdata定義ファイルの格納先フォルダを指定
QST_GMAPDATA_PATH '= "ERB/QUEST/クエストデータ/QUEST_22_ノアレルム辺境の賊徒"
;マップデータ定義XMLとTXTのファイル名を指定。
;2ファイルの名称は拡張子を除いて一致している必要がある
QST_GMAPDATA_FILENAME '= "ノアレルム辺境の賊徒"
;このフラグを1にすると経路描画が行われなくなる。屋内マップなどあえて経路を図示する必要がない場合に立てるとよい
QST_HIDE_PATH = 0

;各エリア（ノード）に配置するイベントIDを記入。0-24はノード番号に対応
;EVENT_LIST末尾の’,’を忘れないように
_DIC_HO_MAPDATA:MAP_ID:0   '= "[EVENT_LIST:0,]"
_DIC_HO_MAPDATA:MAP_ID:1   '= "[EVENT_LIST:0,10,11,]"
_DIC_HO_MAPDATA:MAP_ID:2   '= "[EVENT_LIST:0,20,]"
_DIC_HO_MAPDATA:MAP_ID:3   '= "[EVENT_LIST:0,]"
_DIC_HO_MAPDATA:MAP_ID:4   '= "[EVENT_LIST:0,40,]"
_DIC_HO_MAPDATA:MAP_ID:5   '= "[EVENT_LIST:0,50,]"
_DIC_HO_MAPDATA:MAP_ID:6   '= "[EVENT_LIST:0,60,]"
_DIC_HO_MAPDATA:MAP_ID:7   '= "[EVENT_LIST:0,70,71,]"
_DIC_HO_MAPDATA:MAP_ID:8   '= "[EVENT_LIST:0,80,]"
_DIC_HO_MAPDATA:MAP_ID:9   '= "[EVENT_LIST:0,90,91,92,]"
_DIC_HO_MAPDATA:MAP_ID:10  '= "[EVENT_LIST:0,100,]"
_DIC_HO_MAPDATA:MAP_ID:11  '= "[EVENT_LIST:0,110,111,112,113,114,]"
_DIC_HO_MAPDATA:MAP_ID:12  '= "[EVENT_LIST:0,120,121,]"

;-------------------------------------------------
;クエストのタイプ
;-------------------------------------------------
@QST_TYPE_QUEST_22
RESULTS '= "GMAP"

;-------------------------------------------------
;クエスト固有の初期化処理
;-------------------------------------------------
@QST_INIT_QUEST_22
;クエスト固有のユニークエネミーをエネミー定義データに追加
CALL QST_QUEST_22_ENEMY_DATA
;クエストクールタイムを加算
QST_COOLTIME:QST_有効クエストID += 4

;-------------------------------------------------
; クエスト中止時の処理
; QST_QUEST_{QUEST_ID}_ABORT は撤退コマンド実行時に呼ばれる
;-------------------------------------------------
@QST_QUEST_22_ABORT

;-------------------------------------------------
;クエストの導入文
;-------------------------------------------------
@QST_INTRO_QUEST_22
CALL PRINT_TITLE("ノアレルム辺境の賊徒", "QUEST")
;初回のみ表示
IF QST_SV_QUEST_FLAG:22 == 0
	SETCOLOR 0x8C95EE
	PRINTFORMW 「……情報、取ってきたよ」
	PRINTFORMDW 調査に向かっていたススキが戻ってきた
	PRINTL
	PRINTFORMW （――『草』の者が絡んでいるかもしれないなら、私が調べた方がいいでしょ。蛇の道は蛇、ってね）
	PRINTFORMDW そう言って姿を消してから、さほど時間も経っていない
	PRINTL
	PRINTFORMW 「賊の根城はこの廃集落。大集団ではなく、少ない人数で構成された集団みたい」
	PRINTFORMW 「……とはいえ、相手は『草』の者。罠や伏撃には事欠かないだろうね」
	PRINTL
	PRINTFORMDL 淡々と報告するススキの様子は平時と変わりない
	PRINTFORMDW ……同郷の者を討つことになるかもしれないが大丈夫なのか、と問うと
	PRINTL
	PRINTFORMW 「……ありがと。まあ、こういうことも初めてじゃないし」
	PRINTFORMW 「それに、ちょっと気になることもあってね」
	PRINTL
	PRINTFORMW 「任務で負傷して治療の見込みがなくなった『草』の者は、機密保持のために自死するのが掟、なんだけど……」
	PRINTFORMW 「雇い先との取引の結果、うまく足を洗うヤツもいるわけ。私みたいにね」
	PRINTL
	PRINTFORMW 「でも、いいアガリを迎える者もいれば、降り先を間違える者もいる」
	PRINTFORMW 「今回は、しくじった側……あるいは、もっと悪い事になってるかもしれない」
	PRINTL
	PRINTFORMDW 決然とした表情で、ススキが言葉を切る
	PRINTFORMDW 言外に、最悪の事態の時は容赦なく手を下す、という意思を滲ませて……
	PRINTL
	WAIT
	RESETCOLOR
ENDIF
CALL FL_PLAYBGM("Mountain_0.ogg")
FLAG:クエスト_中断を許可 = 1
;-------------------------------------------------
;イベントリストに登録するイベント情報
;イベントリストに登録するイベント情報を記述する
;<記述方法>
;EVENT_TRIGGER
; "ENTER_ROOM" イベントが配置されたルームの入室時に強制的に実行される
; "EVENT_SLOT" イベントリストのクリック時に実行される
; "TRAIN_EXIT" 調教イベント終了後 ※予め変数:QST_TRAIN_EXIT_EVENT_ID に調教終了後に実行したいイベントIDを指定する必要がある
;EVENT_SLOT_NAME
; イベントリスト上の表示名
;EVENT_TYPE
; "ONCE" クエスト中一度のみ呼出されるイベント(※トリガーが"ENTER_ROOM"のイベントのみ機能)
;           実行済みかをフラグ管理するまでもないイベント(会話など)で使用する想定
;
;戻り値
; RESULT:0 = EVENT_TRIGGER
; RESULT:1 = EVENT_SLOT_NAME
; RESULT:2 = EVENT_TYPE
;-------------------------------------------------
;-------------------------------------------------
;EV0
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_22_EVENT_0
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = ""
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_22_EVENT_0
#DIMS ROOM_NAME

CALL HO_CREAR_LOG()
SELECTCASE HO_MASTER現在地
	CASE 0
		CALL HO_PRINT_LOG("あまり往来のない様子の人道が、少しずつ傾斜を強めていく", 1)
	CASE 1
		CALL HO_PRINT_LOG("よく手入れされた小屋が建っている", 1)
	CASE 2
		CALL HO_PRINT_LOG("山道の脇の小さな窟に、風化の進んだ小さな石造りの祠がある", 1)
	CASE 3
		CALL HO_PRINT_LOG("まったく手入れされていない木々は鬱蒼として先を見通せない…", 1)
		CALL HO_PRINT_LOG("木こりも、ここには踏み入らないようだ", 1)
	CASE 4
		CALL HO_PRINT_LOG("土砂崩れで村道が完全に寸断されている", 1)
		CALL HO_PRINT_LOG("おそらくこれが、この集落の廃れた理由だろう", 1)
		CALL HO_PRINT_LOG("微かな踏み跡が見えるが……羚羊か何かだろうか", 1)
	CASE 5
		CALL HO_PRINT_LOG("密生する藪と天蓋と化した梢の狭間に、わずかな踏み跡が続いている…ように見える", 1)
	CASE 6
		CALL HO_PRINT_LOG("木々の切れ目にガレた急斜面が広がっている", 1)
		CALL HO_PRINT_LOG("これを登るには苦労しそうだ", 1)
	CASE 7
		CALL HO_PRINT_LOG("丘の上に立った……木の枝が払われていれば、眺望も開けただろうに", 1)
	CASE 8
		CALL HO_PRINT_LOG("完全に崩れた家屋の廃材が山となって積みあがっている", 1)
	CASE 9
		CALL HO_PRINT_LOG("屋根の傾いた小屋がある", 1)
		CALL HO_PRINT_LOG("……辛うじて建っている、という風情だ", 1)
	CASE 10
		CALL HO_PRINT_LOG("この集落跡には珍しい、石造りの小屋がある", 1)
		CALL HO_PRINT_LOG("表面が風化してはいるが、堅牢なつくりのようだ", 1)
	CASE 11
		CALL HO_PRINT_LOG("薄暗い地下室は狭く、天井も低い", 1)
		CALL HO_PRINT_LOG("倉庫か、物置として使われていたものだろうか", 1)
	CASE 12
		CALL HO_PRINT_LOG("この集落跡では大きい部類に入る、納屋と思われる建物だ", 1)
		CALL HO_PRINT_LOG("所々の壁が割れ崩れ落ち、屋根葺きも傷んでいる", 1)
		SIF QST_SV_QUEST_FLAG:22 != 2
			CALL HO_PRINT_LOG("しかし、窓などの開口部は全て板で打ち塞がれている…", 1)
ENDSELECT

;-------------------------------------------------
;EV1
;木こりの掘立小屋
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_22_EVENT_10
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%
; ゲーム中一度でも解決済みの場合は非アクティブに
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT
	CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ENDIF

@QST_QUEST_22_EVENT_10
WAIT
PRINTFORMW %QST_LEADER_NAME()%の行く手に、質素なつくりの掘立小屋が見えてくる
PRINTFORMW 小屋の前で、一人の大男が黙々と木挽きの目立てをしている……小屋の主だろうか
PRINTFORMW %QST_LEADER_NAME(1)%が声をかけてみると、男は熊のような仏頂面で応じてきた
PRINTL
PRINTFORMW 「……ここの奥の山？」
PRINTFORMW 「昔は集落があったらしいが、今はあばら家が数軒あるだけの廃村だよ」
PRINTFORMW 「大きな山津波があってな、道どもども大分やられちまったらしい」
PRINTFORMW 「俺も、もうしばらくは立ち入ってねえ……どうも、森によからぬ獣が徘徊し始めてるみたいでな」
PRINTL
PRINTFORMW この辺りに不逞の輩が出入りしていないか、と直裁に尋ねる
PRINTL
PRINTFORMW 「さぁ……そういう手合いにはお目にかかってないな」
PRINTFORMW 「ま、迂闊に足を踏み入れないほうがいい」
PRINTFORMW 「行くなら……道は荒れ放題だが、森の中に獣道くらいは通ってたはずだ」
PRINTL
PRINTFORMW それだけ言い置くと、男は再び作業に戻ってしまった
PRINTFORMW %QST_LEADER_NAME()%は一礼して、先を急ぐことにした…
PRINTL
CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)

;クリア後、再訪すると発生
@QST_ADD_EVENT_QUEST_22_EVENT_11
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

; ゲーム中一度でも解決済みの場合は非アクティブに
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT
	CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ENDIF

@QST_QUEST_22_EVENT_11

IF QST_SV_QUEST_FLAG:QST_有効クエストID == 2
	WAIT
	PRINTFORMW 小屋の前で、木こりが薪割りに勤しんでいる
	PRINTFORMW %QST_LEADER_NAME()%の姿を認めると、眉をぴくり、とさせたが、相も変わらずの仏頂面だ
	PRINTL
	PRINTFORMW 「……あんたらか。集落跡での用事は済んだのか？」
	PRINTFORMW 「……、……そうか。いや、それならいい」
	PRINTL
	PRINTFORMW そう言い置くと、会話は終わりだ、とばかりに、木こりは薪割りに戻った
	PRINTFORMW その背中に、少しばかりの哀愁が透けたのは――まあ、気のせいだろう
	PRINTL
	CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)
ENDIF

;-------------------------------------------------
;EV2
;小さな祠
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_22_EVENT_20
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_22_EVENT_20
WAIT
PRINTFORMW %QST_LEADER_NAME(1)%は、か細い人道の脇に穿たれた小さな窟に目を止めた
PRINTFORMW のぞき込んでみると、小さな石造りの祠が設えられている
PRINTFORMW 道のどん詰まりにあるわりに、よく手入れがされている……先ほどの木こりの仕事だろうか
PRINTFORMW しかし、妙にお供え物が多い気もする
PRINTFORMW ……ちょっとくらい頂戴してもバレないだろうか？
PRINTL
CALL ASK_YN("いただいていく", "放っておきその場を立ち去る")
IF !RESULT
	PRINTFORMW こっそり頂戴していくとしよう
	CALL QST_CHANGE_ITEM(@"{ITEM_パン},2,{ITEM_動物の肉},2,")
ELSE
	PRINTFORMW ……さすがにお供え物に手をつけるのはどうかと思う
	CALL QST_CHANGE_ITEM("名誉点,5,")
ENDIF

;-------------------------------------------------
;EV4
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_22_EVENT_40
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "岩間の湧水"

RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_22_EVENT_40
#DIM EVENT_FLAG
#DIM TARGET_ID

PRINTFORMW %ANAME(QST_RANDCHARA())%は斜面に折り重なった岩々の間から流れ出る湧水を見つけた
PRINTFORMW 水は冷たく澄み、喉の乾きを癒やすにはもってこいだ
PRINTL
PRINTFORML 少し休憩するにはちょうど良さそうに見えるが……
CALL ASK_YN("休憩する", "やめておく")
IF !RESULT
    PRINTFORMW %QST_LEADER_NAME()%は湧水で喉の乾きを癒やした……
    ;パーティ全員の体力を回復
    FOR LOCAL, 0, QST_PARTYNUM
        TARGET_ID = TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))
        CALL QST_CHANGE_HP(TARGET_ID, 1000)
    NEXT
    WAIT
    CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ELSE
ENDIF
CALL QST_SET_EVENT_FLAG(QST_NOWEV, 1)

;-------------------------------------------------
;EV5
;林間の踏み跡
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_22_EVENT_50
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_22_EVENT_50
#DIM  DYNAMIC CHARA_ID
;クリア前の前段、戦闘回避可能
IF QST_SV_QUEST_FLAG:QST_有効クエストID < 2
	WAIT
	PRINTFORMW 森に踏み入った%QST_LEADER_NAME()%は、藪漕ぎと枝くぐりを強いられていた
	PRINTFORMW 手入れされていない鬱蒼とした森はあまり見通しが効かない。注意して進まなくては…
	PRINTL
	;初見殺し…初回のみは判定何度高め
	CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
	IF RESULT
		CALL QST_ROLL_MULTI("観察,", "林間をよく見る(観察),", "200", 5)
	ELSE
		CALL QST_ROLL_MULTI("観察,", "林間をよく見る(観察),", "500", 10)
	ENDIF
	CHARA_ID = RESULT:1
	IF RESULT == 1
		PRINTFORMW %ANAME(CHARA_ID)%は足元に張られた細いテグスの存在に気付いた！
		PRINTW
		PRINTFORMW 何かの罠につながっているのかもしれない
		PRINTFORMW 仲間に警戒を促し、テグスをそっとまたいで通ることにした……
		CALL QST_CHANGE_ITEM("名誉点,20,")
		CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)
		RETURN
	ELSEIF !RESULT
		PRINTFORMW %ANAME(CHARA_ID)%は目を凝らしたが、何も見つけることはできなかった
		PRINTW
	ENDIF
	PRINTFORMW %QST_LEADER_NAME()%が、足に何かがひっかかった感覚を覚えた刹那――
	PRINTFORMW カラカラ、と小高い音が連鎖的に林間に響き渡る
	PRINTL
	PRINTFORMW ――鳴子か、と気づいた時には、聞こえる音は耳をつんざく荒々しい咆哮に取って代わっていた
	PRINTFORMW 伏せていた番犬……というには大きすぎる猛獣が、藪を割って襲い掛かってくる！
;クリア後の前段、強制戦闘
ELSE
	PRINTFORMW %QST_LEADER_NAME()%が鬱蒼とした森を進んでいると、周囲から一斉に遠吠えが響いてきた
	PRINTFORMW どうやら猛獣の縄張りに踏み込んでしまったようだ！
ENDIF
;戦闘イベント
CALL CC_INIT()
;クエストのパーティメンバー全員を攻撃側として戦闘メンバーに追加
CALL CC_ADDPARTY()
;エネミーを防御側で追加
IF QST_SV_QUEST_FLAG:QST_有効クエストID < 2
	CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_22_訓練されたワーグ", "訓練されたワーグ")
ELSE
	CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_22_訓練されたワーグ", "ワーグの長")
ENDIF
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "ワーグ", "ワーグ１", 20)
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "ワーグ", "ワーグ２", 20)
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "ワーグ", "ワーグ３", 20)
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "ワーグ", "ワーグ４", 20)
;戦闘イベント呼出
CALL CLOSE_COMBAT("[BGM:Battle_1.ogg]")
;勝利
IF RESULT == 1
	CALL CC_APPLY_CHARA_HP
	;戦利品獲得処理
	CALL CC_GET_LOOT_FROM("DEFENDER")
	;クリア前のみ文章追加
	IF QST_SV_QUEST_FLAG:QST_有効クエストID < 2
		PRINTL
		PRINTFORMW %ANAME(QST_RANDCHARA())%がしつこく食い下がってくるワーグに一撃を加え、絶命させる
		PRINTFORMW その様子を見たワーグの群れは、怖気づいたのか蜘蛛の子を散らすように逃げ去っていった
		PRINTL
		PRINTFORMW ……ふう、と%QST_LEADER_NAME()%は一息ついた
		PRINTFORMW 今しがた倒したワーグは、妙に戦い慣れしている……手練れ、というに相応しいオーラがあった
		PRINTFORMW やはり、この土地には何者かが潜んでいる
		PRINTFORMW さらに気を引き締めてかかるべきだろう……
		PRINTL
		CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)
		PRINTL
	ENDIF
;敗北
ELSE
	CALL CC_APPLY_CHARA_HP()
	PRINTL
	PRINTFORMW 奇妙なまでに統率されたワーグの動きに翻弄され、傷ばかりが増えていく…
	PRINTFORMW このままでは勝ち目が薄い、戦い続けても命を落とすだけであろう
	PRINTL
	PRINTFORMW %QST_LEADER_NAME()%は包囲を破り、退却することを決意した……！
	;汎用敗北イベントへ
	CALL QST_QUEST_22_DEFEAT_EVENT
ENDIF

;-------------------------------------------------
;EV6
;ガレた急斜面
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_22_EVENT_60
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%
;クエストクリア前は何度でも繰り返し、ただし初見殺しイベとして難易度は下げる
SIF QST_SV_QUEST_FLAG:22 == 2
	CALL QST_DEACTIVATE_EVENT(QST_NOWEV)

@QST_QUEST_22_EVENT_60
#DIM  DYNAMIC CHARA_ID

WAIT
PRINTFORMW %QST_LEADER_NAME()%の目の前には、急傾斜の斜面がそり立っていた
PRINTFORMW さらに斜面は大小さまざまな転石で埋め尽くされており、足場は極めて悪そうだ
PRINTFORMW ここを登り詰めなくては、集落跡にはたどり着けないだろう…
PRINTL

;初見殺し…初回のみは判定難度高め
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT
	CALL QST_ROLL_MULTI("観察,身体,", "斜面をよく見る(観察),斜面を勢いよく駆け登る(身体),", "100,100,", 5)
ELSE
	CALL QST_ROLL_MULTI("観察,身体,", "斜面をよく見る(観察),斜面を勢いよく駆け登る(身体),", "400,500,", 10)
ENDIF
CHARA_ID = RESULT:1
IF RESULT == 1
	PRINTFORMW %ANAME(CHARA_ID)%は、斜面上に不自然に固定された岩石を見つけた
	PRINTFORMW 落石の罠かもしれない、迂回して通ることにしよう……
	PRINTW
	CALL QST_CHANGE_ITEM("名誉点,20,")
	CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)
	RETURN
ELSEIF !RESULT
	PRINTFORMW …………
	PRINTFORMW ………
	PRINTFORMW ……
	PRINTL
	PRINTFORMW ！
	PRINTFORMW 斜面上から一抱えもある岩石が次々と転がってくる！
	PRINTL
	FOR LOCAL, 0, LIST_COUNT(_LIST_QST_PARTY)
		CHARA_ID = TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))
		CALL FL_PLAYSE("Explode_0.mp3")
		IF RAND:(ABL:(CHARA_ID):身体) < 50
			CALL QST_CHANGE_HP(CHARA_ID, -1000, 1)
			WAIT
		ELSE
			PRINTFORMW %ANAME(CHARA_ID)%は辛うじて難を逃れた…
		ENDIF
	NEXT
	PRINTL
ENDIF
CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)

;-------------------------------------------------
;EV7
;廃集落のはずれ
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_22_EVENT_70
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%
; ゲーム中一度でも解決済みの場合は非アクティブに
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT
	CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ENDIF

@QST_QUEST_22_EVENT_70
WAIT
PRINTFORMW なんとか急斜面を登り切った%QST_LEADER_NAME()%の前に、平場が広がっている
PRINTFORMW 山津波の痛々しい傷痕を覆い隠すように、岩々の隙間から緑が繁茂していた
PRINTL
PRINTFORMW 少しずつ森に飲み込まれつつあるそこが、木こりの言った集落跡であろう
PRINTFORMW 遠目には、いくつかのあばら家が見て取れる
PRINTL
PRINTFORMW ……とても、とても静かだ
PRINTFORMW それが、逆に不安感を煽る
PRINTL
PRINTFORMW ここまでの道のりにも、いくつかの周到な罠が張られていた
PRINTFORMW 敵の懐に深く入り込んでいくなら、警戒を怠るべきではないだろう……
PRINTW
CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)

@QST_ADD_EVENT_QUEST_22_EVENT_71
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "白骨化した遺体"

RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_22_EVENT_71

CALL FL_PAUSEBGM()
PRINTFORMW %ANAME(QST_RANDCHARA())%は集落の手前で、白骨化した遺体を見つけた
PRINTFORMW 装備は少し傷んではいるが、それなりにいいものを身につけていたようだ
PRINTL
PRINTFORMW ……上等な傭兵か、あるいは軍の正規兵かもしれない
PRINTFORMW そんなナリをした者がここで息絶えている、という事実が、この先に進めるべき足を重くする
PRINTL
PRINTFORMW 詳細に遺体を検分していて、違和感に気づく
PRINTFORMW ここまで旺盛に芽吹いていた緑が、この遺体のまわりだけぽっかりと更地となっている
PRINTFORMW よく見ると遺体の骨には、他の鳥や獣などの骨も混じっていた
PRINTL
PRINTFORMW ……おそらく、この人間は『毒』にやられたのだろう
PRINTFORMW それも、遺体に残留するレベルのかなり強力なものだ
PRINTFORMW 遺体を漁ったものはことごとくソレに中って命を落とし、同様に白骨を晒すハメになったのだろう
PRINTL
PRINTFORMW ……ずいぶんと厄介なものが控えているようだ
PRINTFORMW 警戒は怠らないようにしよう……
PRINTL
CALL FL_REPLAYBGM
CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
CALL QST_SET_EVENT_FLAG(QST_NOWEV, 1)

;-------------------------------------------------
;EV8
;廃屋の陰
;-------------------------------------------------
; #region
@QST_ADD_EVENT_QUEST_22_EVENT_80
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_22_EVENT_80
CALL FL_STOPBGM()
WAIT
PRINTFORML %QST_LEADER_NAME()%らは足音を潜めて建物の間を進む
PRINTFORMW 森林に半ば飲み込まれかけた廃集落には、鳥の囀りのみが遠く響いている
PRINTL
PRINTFORMW ある崩れた建屋が、その下から芽吹いたツタや若木によって覆いつくされていた
PRINTFORMW めぼしいものはなさそうだが、%QST_LEADER_NAME()%は周囲にしばし目を配った
PRINTL
PRINTFORMW ……？
PRINTFORMW 靴の下にあるモノの違和感に、%ANAME(QST_RANDCHARA())%は目をしばたたかせた
PRINTFORMW 踏みつけたのは、建屋の残骸からはみ出た、ねじくれた樹根だった
PRINTFORMW 何か……踏んだ瞬間、ゴムまりが反発するかのように大きく膨れ上がったような……？
PRINTL
CALL FL_PLAYSE("HeavySmash_1.mp3")
PRINTFORMW 刹那、建屋の残骸がガタガタと震え出す
PRINTFORMW 地震か――と疑ったが、動いているのは目の前の廃屋だけだ！
PRINTL
CALL FL_PLAYSE("Explode_0.mp3")
PRINTFORMW ……ずる、と廃材と緑の小山の中から、巨きなモノが首をもたげた
PRINTFORMW 身もだえるようにその巨体を揺するたびに、バラバラと大小の廃材と破片が落ち、音を立てて地面に降り注ぐ
PRINTFORMW それは、顔を毒々しい大輪の花々にすげ替え、背中にイバラを伴う無数のツタを生やした蛇――大樹のような怪物だ！
PRINTL
PRINTFORMW 周囲を探るようにゆらりと巡らせた怪物の頭花が、ひたと%ANAME(QST_RANDCHARA())%らの方向で止まる
PRINTFORMW ざわざわ、と怪物の全身の葉が一斉にさざめき、繰り返し律動するような葉擦れが不快な音を立てる
PRINTFORMW こちらを敵と認識したのか、図太く発達したツタ――それ自体も小さな蛇のようだ――を展開し、迎撃態勢に入ったようだ
PRINTL
PRINTFORMW 先へ進むには、この怪物を打ち倒すしかないようだ……
PRINTW

;戦闘イベント初期化
CALL CC_INIT()
;クエストのパーティメンバー全員を攻撃側として戦闘メンバーに追加
CALL CC_ADDPARTY()
;エネミーを防御側で追加
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_22_混成栽培兵器６号", "『混成栽培兵器６号』")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_22_茨の蛇蔦", "茨の蛇蔦１")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_22_茨の蛇蔦", "茨の蛇蔦２")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_22_茨の蛇蔦", "茨の蛇蔦３")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_22_茨の蛇蔦", "茨の蛇蔦４")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_22_茨の蛇蔦", "茨の蛇蔦５")
;戦闘イベント呼出
CALL CLOSE_COMBAT("[BGM:Battle_1.ogg][BGM_NOT_AUTOSTOP:1]")

;勝利
IF RESULT == 1
	PRINTL
	PRINTFORMW %QST_LEADER_NAME(1)%は怪物の胴体が地面から伸びていることに目を付けた
	PRINTFORMW アレがもし植物ならば、茎幹を断ち切れば……！
	PRINTL
	PRINTFORMW 鞭のように叩きつけられる巨体を避け、死角を補う蛇蔦の攻撃をもしのぎ、%QST_LEADER_NAME(1)%はその根元に肉薄する
	PRINTFORMW %QST_LEADER_NAME(1)%は得物を全力で打ち付けるが、緻密に木質化した怪物の茎幹には刃があまり徹らない
	PRINTFORMW 一度で駄目なら繰り返し、だ……！
	PRINTL
	PRINTFORMW ……反復、仲間との入れ代わり立ち代わりの攻撃を繰り返すうち、みしり、と何かが軋む音が響く
	CALL FL_PLAYSE("HeavySmash_1.mp3")
	PRINTFORMW めきめきと木質化した茎幹が裂け――大樹を伐り倒すかのように、怪物の巨体が地面に倒れ伏した！
	PRINTL
	CALL FL_STOPBGM()
	;戦闘で増減したステータスをキャラに反映
	CALL CC_APPLY_CHARA_HP()
	;戦利品獲得処理
	CALL QST_CHANGE_ITEM("名誉点,150,")
	CALL CC_GET_LOOT_FROM("DEFENDER")
	PRINTL
	PRINTFORMW %QST_LEADER_NAME()%は打ち倒した怪物を検分するも、その異様さに眉をひそめた
	PRINTFORMW このような巨大で異様な存在は魔物でしかありえない
	PRINTFORMW さりとて、植物型の魔物と言うには、あまりに機能美に欠けている
	PRINTL
	PRINTFORMW まるで、植物に別の生き物を挿げ付けたような……
	PRINTFORMW 妙なざわつきを覚えながら、%QST_LEADER_NAME()%は調査行を再開した
	CALL FL_PLAYBGM("Mountain_0.ogg")
	CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)
;敗北
ELSE
	CALL FL_STOPBGM()
	CALL CC_APPLY_CHARA_HP()
	PRINTL
	PRINTFORMW ――まるで歯が立たない！
	PRINTFORMW 一体どこが弱点なのか、怪物は%QST_LEADER_NAME()%の攻撃を全く意に介していない
	PRINTL
	PRINTFORMW のたうった巨体が、この先へは進ませないという無言の圧をかけてくる……
	PRINTL
	PRINTFORMW %QST_LEADER_NAME()%は命からがら逃げ出した……！
	;ゲームオーバー共通イベント
	CALL QST_QUEST_22_DEFEAT_EVENT
ENDIF
; #endregion

;-------------------------------------------------
;EV12
;大きな納屋
;-------------------------------------------------
; #region
@QST_ADD_EVENT_QUEST_22_EVENT_120
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%
; ゲーム中一度でも解決済みの場合は非アクティブに
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT
	CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ENDIF

@QST_QUEST_22_EVENT_120
WAIT
PRINTFORMW 廃集落を抜けた一同の前に現れたのは、ここまでのどの建屋より大きな納屋だった
PRINTL
PRINTFORMW かつて集落の農作業の基幹を担ったであろう大きな建屋は、人の手を離れてより長く放置されていたようだ
PRINTFORMW 壁はところどころで崩れ、屋根葺きもだいぶ傷んでいる
PRINTL
PRINTFORMW だが……不気味なことに、建屋の全ての窓や戸には板が打ち付けられ、完全に塞がれている
PRINTFORMW 開口部は一切なく、内部の様子はうかがえない
PRINTFORMW 周囲をぐるりと巡ってみるも、入り口は正面の観音開きの大扉のみだ
PRINTL
PRINTFORMW 明らかに、何かが潜んでいる……　踏み込むならば、それ相応の備えをするべきだろう
CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)

@QST_ADD_EVENT_QUEST_22_EVENT_121
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "納屋に踏み入る"
#DIMS EVENT_TYPE = ""
; 派生を検討中
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT == 1
;派生を検討中
;	EVENT_SLOT_NAME '= "焼付いた残留思念"
;ELSEIF RESULT == 2
	CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ENDIF
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_22_EVENT_121
#DIM DYNAMIC CHARA_ID

;ススキ以外のメンバーを選出（テキストの整合性担保のため）
FOR LOCAL, 0, LIST_COUNT(_LIST_QST_PARTY)
	CHARA_ID = TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))
	SIF CHARA_ID == NAME_TO_CHARA("UC_ススキ")
		CONTINUE
	BREAK
NEXT

CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT != 1
	CALL FL_STOPBGM()
	CALL FL_PLAYSE("Woodcut_0.mp3")
	WAIT
	PRINTFORMW %QST_LEADER_NAME()%らは納屋の中に踏み入った
	CALL FL_PLAYBGM("Ambient_dark_2.mp3")
	PRINTFORMW 陽光が一切入り込まない建屋の中は暗く、内部の様子はよく見通せない……
	PRINTFORMW ただ、風船から空気が抜けるような、細く長い空気の噴出音のみが繰り返し聞こえてくる
	PRINTL
	PRINTFORMW ……
	PRINTFORMW ………
	PRINTFORMW 目が慣れてくると、そこに蹲る怪異に目が釘付けになった
	PRINTL
	PRINTFORMW 大きな納屋の闇の中央に、大きな三角錐の形をした塊が鎮座している
	PRINTFORMW その表面はごわごわした硬質な皮膚と、まばらに生えた鱗に覆われている
	PRINTFORMW 納屋の土間を這うように、数本の触腕らしきものがのたくっているのが見えた
	PRINTL
	PRINTFORMW 体表から不規則に突き出た４つの小さな突起のように見えたものは……
	PRINTFORMW 見間違えでなければ、しごくまっとうな人間の腕と、足のようだ
	PRINTL
	PRINTFORMW やもすると見落としそうになった――その小山のたもとに、人型の何者かが跪いている
	PRINTL
	PRINTFORMW 「……きタ、カ」
	PRINTL
	PRINTFORMW 静かに紡がれた声は、確かに人語のそれに聞こえた
	CALL FL_PLAYSE("Pat_3.mp3")
	PRINTFORMW しかし、ぎくしゃくと立ち上がった大柄な身体のシルエットは、まっとうな人間の形をしていない……！
	PRINTL
	PRINTFORMW 「ここ、で、ヤラれ、るワけにハ、いカナい」
	PRINTL
	PRINTFORMW 発声器が上手く機能していないのか、喘鳴のようにとひどく聞き取りづらい声
	PRINTFORMW ぬばたまの闇から、ぬう、と現れたその姿もまた――まさしく、異形という他なかった
	PRINTL
	PRINTFORMW 顔を覆う面頬を装備した人間の胴体から伸びていたのは、黒光りする剛毛に覆われた図太い魔獣の爪腕
	PRINTFORMW 鱗に鎧われた鳥のようなシャープな逆関節の脚が２本、腰に挿げ付けられている
	PRINTL
	PRINTFORMW 見てくれはどうみても異形だが、その所作は極めて人間のそれに近い
	PRINTFORMW %QST_LEADER_NAME()%らは困惑する……この生き物は、一体なんだ？
	PRINTL
	PRINTFORMW こちらを値踏みするような視線を投げていた異形が、ある一点で目を止める
	PRINTFORMW 視線の先には……
	PRINTL
	PRINTFORMW 「！　……おマえ、ハ……『草』の者、カ」
	SETCOLOR 0x8C95EE
	PRINTFORMW 「……いかにも」
	PRINTL
	CALL FL_PLAYSE("Unlock_0.mp3")
	PRINTFORMDW ススキはその視線に応じて、すらりと得物を抜き放つ
	PRINTFORMDW その様を眩しそうに見届けて……異形が、ゆらり、と構えを取る
	PRINTFORMDW 奇しくも――双方鏡写しのように、同じ構えだった
	PRINTL
	PRINTFORMW 「介錯してやる――辞世の句でも詠むといい」
	PRINTFORMDW 「く、カカッ……　死ねナい、マダ、死ねナいぞ！　■■■―――！！」
	PRINTL
	CALL FL_PLAYSE("Speed_0.mp3")
	WAIT
	CALL FL_PLAYSE("Throw_0.mp3")
	PRINTFORMDW 異形の跳躍はその身をやすやすと天井に運び、それを追ってススキが宙に跳び上がる
	PRINTFORMDW 咆哮する異形の、もはや判別不能な言霊は、誰に宛てたものだったのか――
	PRINTL
	PRINTFORMDW 同時、納屋の中央に鎮座していた肉塊が興奮したように激しく律動を始める
	PRINTFORMDW ぎらり、と大小無数の眼が次々に見開かれ、こちらに一斉に据えられた
	PRINTL
	PRINTFORMDW 「A――Ahaaaaaa！！」
	PRINTL
	CALL FL_PLAYSE("HeavySmash_0.mp3")
	PRINTFORMDW 長く図太い触腕が一斉に持ち上がり、地面を叩いて怪異が動き出した
	PRINTFORMDW その巨体とは不釣り合いなほどの運動能力で、跳ね飛びながら砲弾のようにこちらに飛び込んでくる！
	PRINTL
	PRINTFORMDW ここでこの異形どもを止めなくては……
	PRINTW
	RESETCOLOR
	;戦闘イベント初期化
	CALL CC_INIT()
	;クエストのパーティメンバー全員を攻撃側として戦闘メンバーに追加
	CALL CC_ADDPARTY()
	;エネミーを防御側で追加
	CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_22_改造忍者・鳥兜", "『改造忍者・鳥兜』")
	CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_22_被験体３号", "『被験体３号』")
	CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_22_訓練されたワーグ", "訓練されたワーグ１")
	CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_22_訓練されたワーグ", "訓練されたワーグ２")
	CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_22_訓練されたワーグ", "訓練されたワーグ３")
	;戦闘イベント呼出
	CALL CLOSE_COMBAT("[BGM:Boss_0.ogg][BGM_NOT_AUTOSTOP:1]")

	;勝利
	IF RESULT == 1
		SETCOLOR 0x8C95EE
		PRINTL
		PRINTFORMW 「ふ……ッ！！」
		PRINTL
		PRINTFORMDW ススキは小刻みに気を吐きながら、異形の忍者との凄絶な攻撃の応酬を繰り返す
		SELECTCASE ABL:NAME_TO_CHARA("UC_ススキ"):戦闘
			CASE IS < 100
				PRINTFORMDW 五分五分に立ち回っているようだが……実際のところは何とか凌いでいる、という状況だ
			CASE IS < 150
				PRINTFORMDW 五分五分に立ち回ってはいるが、いつまでもつものか
			CASEELSE
				PRINTFORMDW ややススキが押しているように見えるが、それも寸毫の差でしかない
		ENDSELECT
		PRINTL
		PRINTFORMDW 異形と化した忍者の『技』と『術』は、おそらく人だった頃と変わらぬ精妙の極みを見せ――
		PRINTFORMDW その持ち前の技巧に、魔物の膂力と慮外の身体駆動が乗せられる！
		PRINTFORMDW その結果は――
		PRINTL
		PRINTFORML 「……ッ！？」
		PRINTFORMDW 「カッ――クカカカッ！！」
		PRINTL
		CALL FL_PLAYSE("Ice_0.mp3")
		PRINTFORMDW 毒霧を目晦ましに、異形の忍者の背中に隠された３本目・４本目の腕が展開――ススキに殺到する！
		PRINTL
		RESETCOLOR
		PRINTFORMDW …………
		PRINTFORMDW ………
		PRINTFORMDW ……
		PRINTL
		PRINTFORMW 巨体を揺すりながら突撃してくる怪異は、その大量の眼で全周囲をくまなく索敵しているようだった
		PRINTFORMW 包囲しても上手く躱され、%ANAME(CHARA_ID)%らは数の優位を活かせない
		PRINTFORMW 埒が明かない……そう思った刹那、%ANAME(CHARA_ID)%に振り回された触腕が迫る
		PRINTL
		PRINTFORMW 咄嗟、%ANAME(CHARA_ID)%は拝み打ちに触腕を一太刀に破断し――
		CALL FL_PLAYSE("HeavySmash_3.mp3")
		PRINTFORMW 斬り飛ばされた触腕が、窓に打ち付けられた板を弾き飛ばし、建屋外に飛び出していった
		PRINTL
		PRINTFORMW 「Ah―――！？」
		PRINTL
		PRINTFORMW 窓からの陽光が屋内に射し込み――怪異がわずかに後ずさった、ように見えた
		PRINTFORMW はたと、この納屋の開口部が全て塞がれていることに思い当たる
		PRINTFORMW あれだけ目が多いなら――直射を目で浴びるのは苦手なのでは？
		PRINTFORMW 即座に%ANAME(CHARA_ID)%は仲間に指示を飛ばし、板で閉ざされた戸や窓を片っ端から打ち破らせた！
		PRINTL
		PRINTFORMW 「Ah――Gruu―――」
		PRINTL
		PRINTFORMW 眼をつむり、触腕をも使って影を作り……あからさまに怪異の動きが鈍る
		PRINTFORMW 今が、絶好の機だ！
		PRINTL
		PRINTFORMW 一斉に攻撃に転じた%ANAME(CHARA_ID)%らの渾身の一撃が、怪異に殺到した
		PRINTFORMW 怪異の悲鳴めいた絶叫が納屋に響き渡る――
		PRINTL
		PRINTFORMW …………
		PRINTFORMW ………
		PRINTFORMW ……
		PRINTL
		CALL FL_STOPBGM()
		CALL FL_PLAYSE("Hit_0.mp3")
		PRINTFORMW 「！　――しくじっ、タ」
		PRINTL
		PRINTFORMW 肺腑からの血と共に零れた声……白刃が、深々と異形の忍者の背中に突き刺さる
		PRINTFORMW 隠し爪腕の奇襲を受けたはずのススキの姿は靄のように解け、いつの間か背後に入り身しているではないか
		PRINTFORMW 分身の術による幻惑――見破られていたら通用しない手ではあった、が……
		PRINTL
		SETCOLOR 0x8C95EE
		PRINTFORMW 「アンタ……戦いの最中に、よそ見したね？」
		PRINTL
		PRINTFORMDW ぎちり、と臓腑を掻き回すように刃を返したススキが、背後から冷たく囁き――
		PRINTL
		PRINTFORMW 「是にて――御免」
		PRINTL
		CALL FL_PLAYSE("HeavySlash_0.mp3")
		PRINTFORMDW 両手で刃を旋回一閃――ススキは、一刀のもと異形の脇腹から半身を抉り裂いた
		CALL FL_PLAYSE("HeavySmash_1.mp3")
		PRINTFORMDW どう、と地に落ち、痙攣した異形の身体の四肢が、見る見るうちに萎れていく
		CALL FL_PLAYSE("Slash_4.mp3")
		PRINTFORMDW カン、カン……と小高い音を立てて、顔から剥がれた面頬が地面に跳ね飛んだ………
		PRINTL
		CALL FL_PLAYSE("Fanfare_0.ogg")
		PRINTFORMDW ――『ノアレルム辺境の賊徒』を討伐した！
		PRINTL
		CALL FL_PLAYBGM("Mountain_0.ogg")
		PRINTFORMDW …………
		PRINTFORMDW ………
		PRINTFORMDW ……
		PRINTL
		PRINTFORMDW %ANAME(CHARA_ID)%は、座り込んでいたススキにねぎらいの言葉をかける
		PRINTL
		PRINTFORMW 「……ありがと。そっちが上手くやってくれてなかったら、やられてたのは私のほうだった」
		PRINTL
		PRINTFORMDW ススキは%ANAME(CHARA_ID)%の手を取り、ゆっくりと立ち上がった
		PRINTL
		PRINTFORMW 「デカブツの断末魔の悲鳴を聞いて、露骨に隙ができたから」
		PRINTFORMW 「もしかしたら……守りたかったのかも、ね」
		PRINTL
		PRINTFORMDW 重なるように斃れ伏した二体の骸に思うところもあるのか、ススキはじっと彼らの亡骸を見つめていた
		PRINTFORMDW ……と、ふいにススキが、つかつかと納屋の隅に歩いていき、黒光りする何かを拾い上げた
		PRINTFORMDW 先刻まで人型の異形がつけていた、顔の下半分を覆う漆黒の面頬だ
		PRINTL
		PRINTFORMW 「この面頬、もらってこっか。三途の川の渡し賃、六文銭の代わりってことで――」
		PRINTL
		PRINTFORMDW 言葉とは裏腹に、ススキは面頬を丁重に捧げ持つ
		PRINTFORMDW その姿はまるで、異郷の地で果てた同胞を弔うように見えた――
		PRINTL
			{
				CALL QST_GET_WEAPON(
					8
					,"漆黒の面頬"
					,WPN_材質_銀
					,3
					,"[WPN_DECOY_AMPLIFIER:2]"
					,"[WPN_FRONT_DEFENCE:2]"
					,WPN_装備種別_レリック
					,1
					)
			}
		;戦闘で増減したステータスをキャラに反映
		CALL CC_APPLY_CHARA_HP()
		;戦利品獲得処理
		CALL CC_GET_LOOT_FROM("DEFENDER")
		CALL QST_CHANGE_ITEM("MONEY,20000,名誉点,300,")
		CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)
		CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
		CALL QST_QUEST_22_ENDING
	;敗北
	ELSE
		CALL CC_APPLY_CHARA_HP()
		PRINTL
		PRINTFORMW 跳ね飛ぶ大きな怪異と、縦横無尽に飛び回る異形の忍者
		PRINTFORMW 大小の魔の織り成す奇妙な程に息の合った連携に、形勢不利は否めない……！
		PRINTL
		PRINTFORMW 先に返り討ちにあった連中の覆轍を踏むわけにはいかない
		PRINTFORMW ギリギリの情勢下で、%QST_LEADER_NAME()%は退却を決意した
		PRINTL
		PRINTFORMW 追い散らすような追撃がそう長くは続かなかったことに違和感をもつ余裕もない
		PRINTFORMW %QST_LEADER_NAME()%は這う這うの体で、集落から森へと落ち延びるのだった
		PRINTL
		;ゲームオーバー共通イベント
		CALL QST_QUEST_22_DEFEAT_EVENT
	ENDIF
ELSE

ENDIF

; #endregion

;-------------------------------------------------
;EV10
;古びた石造りの小屋
;-------------------------------------------------
; #region
@QST_ADD_EVENT_QUEST_22_EVENT_100
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = ""
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%
; ゲーム中一度でも解決済みの場合は非アクティブに
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT
	CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ENDIF

@QST_QUEST_22_EVENT_100
#DIM DYNAMIC CHARA_ID
WAIT
PRINTFORMW %QST_LEADER_NAME()%は、集落の端の一件の建物の前で足を止めた
PRINTL
PRINTFORMW この集落の中では珍しい、石造りの小屋
PRINTFORMW 物置がわりに使われていたものか
PRINTFORMW 屋内には目ぼしいものは見当たらないが……
CALL QST_ROLL_MULTI("観察,", "小屋の中をさらに調べる(観察),", "75,", , ,"見切りをつける",)
CHARA_ID = RESULT:1
IF RESULT == 1
	CALL FL_PLAYSE("Mining_0.mp3")
	PRINTFORMW %ANAME(CHARA_ID)%は、石畳の下に隠された小さな階段を見出した！
	PRINTL
	PRINTFORMW どうやら、地下室と思しき空間につながっているようだ…
	PRINTL
	CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)
	CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ELSEIF !RESULT
	PRINTFORMW %ANAME(CHARA_ID)%は目を凝らしたが、何も見つけることはできなかった
	PRINTL
	;探索が成功しない場合は地下室へは移動できない
	;廃屋の陰(8)へ強制的に戻す
	CALL HO_MOVE_MASTER(8)
	CALL QST_SET_FOG_OF_WAR(11)
ELSEIF RESULT == -1
	CALL HO_MOVE_MASTER(8)
	CALL QST_SET_FOG_OF_WAR(11)
ENDIF

;-------------------------------------------------
;EV11
;隠された地下室
;-------------------------------------------------
; #region
@QST_ADD_EVENT_QUEST_22_EVENT_110
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%
; ゲーム中一度でも解決済みの場合は非アクティブに
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT
    CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ENDIF

@QST_QUEST_22_EVENT_110

WAIT
PRINTFORMW %QST_LEADER_NAME()%が秘された階段を下っていくと、すぐに小さな地下室に行き当たった
PRINTL
PRINTFORMW 暗闇の中に、一抱え程の大きさの木櫃が置かれている
PRINTFORMW 罠も鍵もかかっていないようだ
PRINTFORMW 蓋を開けると、ノートや書類の束が小奇麗に収められている
PRINTFORMW 誰かが整理してまとめたもののようだ……
PRINTW
CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)

@QST_ADD_EVENT_QUEST_22_EVENT_111
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "『くたびれた研究ノート』"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_22_EVENT_111

CALL FL_PAUSEBGM()
PRINTFORML 何者かの研究ノートのようだ
$INPUT_LOOP
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
PRINTL
CALL ASK_MULTI_JUDGE_HIDE_DEACTIVE("『日焼けした端書き』", 1, "『付箋１―移植』",1, "『付箋２－改造人間』", 1, "『通知書』", 1, "『最後の頁』", RESULT, "読むのをやめる", 1)
SELECTCASE RESULT
	CASE 0
		CALL FL_PLAYSE("Page_0.mp3")
		CALL COLOR_LINE()
		PRINTFORMW 『日焼けした端書き』
		CALL COLOR_LINE()
		PRINTL
		PRINTFORMW 先の大戦以降、魔物の大侵攻は止まったが、いつまた我ら人類に牙を剥くか知れたものではない
		PRINTFORMW 魔物どもを鏖殺するためには、最早手段は選んでいられない
		PRINTL
		PRINTFORMW 人外のものどもを相手にするならば、必然、我々も人の枠を超えねばならない
		PRINTFORMW 結論は自明である
		PRINTL
		SETCOLOR カラー_赤
		PRINTFORMW 魔物を殺すには――
		PRINTFORMW その魔物に倣うべきであろう
		RESETCOLOR
		PRINTL
		PRINTFORMW 　　　　　　　　　　　　　　　　　　　　　　主任研究員　ドクター・フォー　記す
		PRINTW
	CASE 1
		CALL FL_PLAYSE("Page_0.mp3")
		CALL COLOR_LINE()
		PRINTFORMW 『付箋１―移植』
		CALL COLOR_LINE()
		PRINTL
		PRINTFORMW 魔物の細胞組織（以下、M組織と呼称）は、経口摂取や組織への付着では他生物に何の影響ももたらさない
		PRINTFORMW しかし、ある処理を行ってから別の生体組織と結合させると、わずかな確率で融合……定着する
		PRINTL
		PRINTFORMW さりとて、M細胞小片からの増殖を待つのでは時間がかかり過ぎる
		PRINTFORMW やはり、M組織の一部を既存の生命体への接合する処理が、最もスジのよい方法だろう
		PRINT
		PRINTFORMW 既に植物体へのM組織の定着試験を行い、そこから採取した種子の変異体への成長を確認している
		PRINTFORMW これは戦線に播種するだけで展開可能な、疑似的な防衛兵器として運用することができるだろう
		PRINTL
		PRINTFORMW 次はいよいよ動物を対象とする実験に入る
		PRINTFORMW 研究を、急がねばならない……
		PRINTW
	CASE 2
		CALL FL_PLAYSE("Page_0.mp3")
		CALL COLOR_LINE()
		PRINTFORMW 『付箋２－改造人間』
		CALL COLOR_LINE()
		PRINTL
		PRINTFORMW 知性のない動物への移植では、魔物を一体増やすだけに留まってしまう
		PRINTFORMW 兵器として運用するのならば、やはり知性体……人間へのM組織の定着を目指すべきであろう
		PRINTFORMW M組織による『改造人間』計画を、上層部に具申することとする
		PRINTL
		PRINTFORMW 不幸中の幸いか、リナシメントのおかげで融通の利く傷病兵や、血の気に逸る虜囚は数多くいる
		PRINTFORMW ……それに、良い検体も手に入った
		PRINTFORMW 間諜に侵入してきた特殊技術を身につけた屈強な兵士
		PRINTL
		PRINTFORMW あれを被験体とし、我が『改造人間』の有用性を実証してみせよう
		PRINTW
	CASE 3
		CALL FL_PLAYSE("Page_0.mp3")
		CALL COLOR_LINE()
		PRINTFORMW 『通知書』
		CALL COLOR_LINE()
		PRINTL
		PRINTFORMW 被験体が暴走し、所長を含む研究施設の人命に多大な損害が発生
		PRINTFORMW 状況９－４発生に伴い、制圧部隊の派遣を行うものとする
		PRINTL
		PRINTFORMW なお、当研究所の閉鎖、および研究資料の隠滅を並行して進める
		PRINTFORMW 生存者の身柄は公安部預かりとし、監視下での謹慎を命じるものとする
		PRINTW
		CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)
	CASE 4
		CALL FL_PLAYSE("Page_0.mp3")
		CALL COLOR_LINE()
		PRINTFORMW 『最後の頁』
		CALL COLOR_LINE()
		PRINTL
		PRINTFORMW （……乱雑な筆致で、慌てて書きつけられたもののと見える）
		PRINTL
		PRINTFORMW 今回は失敗だった
		PRINTFORMW ヒトより変じたものも、所詮は魔物であった
		PRINTL
		PRINTFORMW だが、貴重かつ膨大な臨床データを得ることができた
		PRINTFORMW 次は　次こそは　もっと上手くやってみせる
		PRINTL
		PRINTFORMW 『改造人間』計画は終わらない
		PRINTFORMW 全ての魔物を駆逐する　その日まで
		PRINTW
	CASEELSE
		CALL FL_REPLAYBGM
		RETURN
ENDSELECT
GOTO INPUT_LOOP

@QST_ADD_EVENT_QUEST_22_EVENT_112
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "『新しめの手帳』"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_22_EVENT_112

CALL FL_PAUSEBGM()
PRINTFORMW さほど傷んでいない手帳があった
PRINTL
PRINTL
CALL FL_PLAYSE("Page_0.mp3")
CALL COLOR_LINE()
PRINTFORMW 昨今の研究室の様子
CALL COLOR_LINE()
PRINTL
PRINTFORMW 博士はここの所、何かにとり憑かれたように研究に没頭している
PRINTFORMW それはとても痛ましく、それと同じくらい、とても……おぞましい
PRINTL
PRINTFORML 寝食を疎かにしているせいで身体は痩せさらばえているというのに、
PRINTFORMW 見開いた目は爛々と輝き、異様なほどに生気が漲っている
PRINTFORMW まるで、博士自身が■■を宿したかのようだ
PRINTL
PRINTFORMW その『成果物』もまた、とびきりの異様さだ
PRINTFORML ２号房の被験体のように、人間の姿かたちに、醜悪な四肢がついているくらいならまだマシな部類である
PRINTFORMW （それが人語を操るのだ、頭がおかしくなりそうだ！）
PRINTL
PRINTFORMW ……３号房の被験体はどうだ
PRINTFORMW 触腕を生やし、ぶくぶくと膨れ上がった肉塊――と思しきものが、部屋を埋めるように鎮座している
PRINTFORMW これが元々は人間の姿をしていたとは　自分が経過観察をしているのでなければ、一笑に付しているところだ
PRINTL
PRINTFORMW ３号被験体が仮死状態にあるか、植物化してくれているのならまだよかったのだ
PRINTFORMW 思い出したくもないが……物音に反応して、ぎょろり、と体に不釣り合いなほどに大きな眼球が動く
PRINTFORMW その太い視線を、こちらにひたと据えてくるのだからたまらない
PRINTL
PRINTFORMW ……ほんとうに、気が狂いそうだ
PRINTFORMW 博士に話しかけるのもとても気が重いが、担当を変えてもらえないか談判をするつもりだ…
PRINTW
CALL FL_REPLAYBGM
CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)

@QST_ADD_EVENT_QUEST_22_EVENT_113
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "『封を切られた手紙』"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_22_EVENT_113

CALL FL_PAUSEBGM()
PRINTFORMW 封を切られた手紙があった
PRINTL
PRINTL
CALL FL_PLAYSE("Page_1.mp3")
PRINTFORMW ……あの男は危険だ。神にでもなったつもりなのか
PRINTFORMW まったく常軌を逸している、魔物の身体の一部を、人間に移植するなどと
PRINTL
PRINTFORMW 画期的な成果を出してこそいるが、行われていることは悪鬼悪魔の所業だ
PRINTFORMW 人体実験を繰り返し、廃人と化してもまるで悪びれず、次の検体を要求してくる始末
PRINTL
PRINTFORMW もし他国にこのような研究が行われていると知れれば、批判は免れないだろう
PRINTFORMW ……いつまでも隠しおおせるものではない
PRINTFORMW あの男は、すぐにでもこの研究所から排除するべきだ
PRINTL
PRINTFORMW そのためには、あれのシンパとなっている有力者との提携を切り崩す必要がある
PRINTFORMW 大臣との個人的なツテを頼って、働きかけを依頼することにする……
PRINTL
PRINTFORMW 　　　　　　　　　　　　　　　　　　国立■■研究所　所長　■■■■■■■
PRINTW
CALL FL_REPLAYBGM
CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)

@QST_ADD_EVENT_QUEST_22_EVENT_114
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "『雑紙』"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_22_EVENT_114

CALL FL_PAUSEBGM()
PRINTFORMW 一見するとただの雑紙だが……
PRINTFORMW ミミズのような、ひどく乱れた筆致で、なにかが書きつけられている
PRINTFORMW 全てを解読するのは困難極まりない
PRINTL
PRINTL
CALL FL_PLAYSE("Page_0.mp3")
SETCOLOR COLOR("シルバー")
;前後の記憶がひどくあやふやになってきている
PRINTFORMW 前後の記憶■ひど■あやふや■■▲てきて■る
;今はいつだ？何日経ったのだ
PRINTFORMW 今■いつ■？　何●経▲た■■
PRINTL
;あのヤブ医者の手術後の副作用は酷いものだ
PRINTFORMW あのヤブ●●の手術後■副作用■酷い■■■
;しかし、ようやく身体の自由が利くようになってきた
PRINTFORMW しかし、よ■■く身体■●由が利く■■■な▲てきた
PRINTL
;任務はもう捨ておくことにする
PRINTFORMW 任●は■■捨てお■■とにす■
;それよりも隣の房の少女を救わなくては
PRINTFORMW それよ■も　隣の房の少女■救わな■ては
PRINTL
PRINTL
PRINTFORMW 私■こんな体になっ■も、笑顔で接してくれた、彼女を
RESETCOLOR
PRINTL
PRINTL
PRINTFORMW ……きれぎれの内容だ。何かを判断するには、少々心許ないだろう
PRINTW
CALL FL_REPLAYBGM
CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)
; #endregion

;-------------------------------------------------
;EV90~ 崩れかけた小屋
;宝箱(一回限り)
;-------------------------------------------------
;イベント登録
; #region
@QST_ADD_EVENT_QUEST_22_EVENT_90
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME

EVENT_SLOT_NAME '= @"宝箱"
;取得済みの場合は非アクティブに
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT
    CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ENDIF
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

;イベント実行内容
@QST_QUEST_22_EVENT_90
#DIMS DYNAMIC _ITEM_LIST

_ITEM_LIST '= @"{ITEM_銀インゴット},3,"
CALL QST_CEV_LOCKED_CHEST(_ITEM_LIST)
IF RESULT
	CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)
ENDIF

;イベント登録
@QST_ADD_EVENT_QUEST_22_EVENT_91
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME

EVENT_SLOT_NAME '= @"木箱"
;取得済みの場合は非アクティブに
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT
    CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ENDIF
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

;イベント実行内容
@QST_QUEST_22_EVENT_91
#DIMS DYNAMIC _ITEM_LIST

_ITEM_LIST '= @"{ITEM_小麦},7,"
CALL QST_CEV_CHEST(_ITEM_LIST)
IF RESULT
    CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)
ENDIF

;イベント登録
@QST_ADD_EVENT_QUEST_22_EVENT_92
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "柳行李"
;取得済みの場合は非アクティブに
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT
    CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ENDIF
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

;イベント実行内容
@QST_QUEST_22_EVENT_92

CALL FL_PLAYSE("Dig_1.mp3")
PRINTFORMW 柳行李の中から略奪品と思われる衣類を見つけた
CALL CHANGE_CLOTH_STOCK(衣装部位_上下服, 9, 1)
CALL CHANGE_CLOTH_STOCK(衣装部位_首, 14, 1)
CALL CHANGE_CLOTH_STOCK(衣装部位_上下下着, 7, 1)
CALL CHANGE_CLOTH_STOCK(衣装部位_靴, 3, 1)
CALL PRINT_NOTICE("衣服:華国のドレスを手に入れた")
CALL PRINT_NOTICE("衣服:ペンダントを手に入れた")
CALL PRINT_NOTICE("衣服:ネグリジェを手に入れた")
CALL PRINT_NOTICE("衣服:サンダルを手に入れた")
PRINTW
CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)
CALL QST_DEACTIVATE_EVENT(QST_NOWEV)

; #endregion

;-------------------------------------------------
;以下イベントリストで管理する必要のない処理,イベント固有の関数など
;-------------------------------------------------
;汎用敗北イベント
@QST_QUEST_22_DEFEAT_EVENT
PRINTL
PRINTFORMW ……………………
PRINTFORMW ………………
PRINTFORMW …………
PRINTL

PRINTFORMW %QST_LEADER_NAME()%はボロボロの身を引きずってなんとか野営地まで帰還した……
PRINTFORML
PRINTFORMW ……今回の探索は失敗に終わってしまった。帰還して一度態勢を立て直すことにしよう
PRINTL
CALL PRINT_NOTICE("治療費や宿泊費として15,000ディナリを失った……")
MONEY -= 15000
QST_SV_QUEST_FLAG:QST_有効クエストID = 1
TFLAG:クエスト_完了フラグ = 1
CALL TRAIN_EXIT()

;-------------------------------------------------
;以下イベントリストで管理する必要のない処理,イベント固有の関数など
;-------------------------------------------------
@QST_QUEST_22_ENDING
PRINTL
PRINTFORMW ……………………
PRINTFORMW ………………
PRINTFORMW …………
PRINTL
CALL COLOR_PRINTW(@"―― %QST_LEADER_NAME()%はノアレルム辺境集落跡を踏破した！", カラー_注意)
QST_SV_QUEST_FLAG:QST_有効クエストID = 2
PRINTFORML
PRINTFORMW 望むならこのまま探索を続けてもよい。疲労が濃いならばここで探索を終了してもよい
PRINTL
CALL ASK_YN("探索を続ける", "帰還する")
PRINTL
IF RESULT
    TFLAG:クエスト_完了フラグ = 1
    CALL TRAIN_EXIT()
ENDIF

;-------------------------------------------------
;@QST_QUEST_22_ENEMY_DATA
; クエスト固有エネミーの読み込み処理
; クエストで共通的に使い回すエネミーの戦闘データは@CC_ENEMY_DATAに記述されているが
; クエスト側で後付で追加することもできる。(@QST_INIT_QUEST_XXなどで明示的に読込み処理を呼び出すこと)
; 特定のクエストでしか使わないエネミーのデータ（ボスなど）はクエスト側に記述するとよいだろう
; パラメータの詳細は @CC_ENEMY_DATA 参照
; ！！！ENEMY_DEF_NAMEはユニークである必要があるためクエスト側でエネミーを追加する場合はENEMY_DEF_NAMEに'QUEST_XX_'を接頭すること！！！
;-------------------------------------------------
@QST_QUEST_22_ENEMY_DATA
{
    DT_ROW_ADD "ENEMY_DATA"
        ,"ENEMY_DEF_NAME", "QUEST_22_改造忍者・鳥兜"
        ,"MAX_HP", 8500
        ,"STR", 145
        ,"MAG", 35
        ,"ATTACK", 145
        ,"DEFENCE", 35
        ,"ARMOR", 250
        ,"AGILITY", 180
        ,"HATE_RATE", 80
        ,"TARGETING", WPN_脅威選定_魔術師
        ,"ENEMY_AI", "[SKILL_USE_RATE:55]"
        ,"REGIST_POISON", 100
        ,"REGIST_STUN", 10
        ,"REGIST_CHARM", 70
        ,"REGIST_PANIC", 70
        ,"REGIST_BLIND", 70
        ,"REGIST_DEATH", 100
        ,"ENEMY_PASSIVE_SKILL", "[CHASE_POISON:3][ATK_ADD_POISON:3]"
        ,"ENEMY_ACTIVE_SKILL", "[ACT_ENEMY_SHINOBI_POISON:2][ACT_REACT:3]"
        ,"DROP_ITEM_LIST", @"MONEY,3000,1000,{ITEM_技巧の宝玉},1,1000,{ITEM_魔獣の肉片},3,1000,"
        ,"EXP", 12
}
{
    DT_ROW_ADD "ENEMY_DATA"
        ,"ENEMY_DEF_NAME", "QUEST_22_被験体３号"
        ,"MAX_HP", 18000
        ,"STR", 160
        ,"MAG", 40
        ,"ATTACK", 160
        ,"DEFENCE", 40
        ,"ARMOR", 400
        ,"AGILITY", 70
        ,"HATE_RATE", 120
        ,"TARGETING", WPN_脅威選定_無作為
        ,"ENEMY_AI", "[SKILL_USE_RATE:35]"
        ,"REGIST_POISON", 50
        ,"REGIST_STUN", 10
        ,"REGIST_CHARM", 70
        ,"REGIST_PANIC", 70
        ,"REGIST_BLIND", 70
        ,"REGIST_DEATH", 100
        ,"ENEMY_ACTIVE_SKILL", "[SMITE:3][ACT_FRONT_DEFENCE:3]"
        ,"DROP_ITEM_LIST", @"MONEY,2500,1000,{ITEM_技巧の宝玉},1,1000,{ITEM_魔獣の心核},1,1000,"
        ,"EXP", 12
}
{
    DT_ROW_ADD "ENEMY_DATA"
        ,"ENEMY_DEF_NAME", "QUEST_22_訓練されたワーグ"
        ,"MAX_HP", 5000
        ,"STR", 60
        ,"MAG", 25
        ,"ATTACK", 80
        ,"DEFENCE", 25
        ,"ARMOR", 150
        ,"AGILITY", 130
        ,"HATE_RATE", 100
        ,"TARGETING", WPN_脅威選定_低耐久
        ,"ENEMY_ACTIVE_SKILL", "[SMITE:2][STUN_ROAR:1][ACT_TORN_CLOTH:1]"
        ,"DROP_ITEM_LIST", @"MONEY,1500,1000,{ITEM_動物の肉},1-3,450,{ITEM_魔獣の肉片},1-3,300,"
        ,"EXP", 8
}
{
    DT_ROW_ADD "ENEMY_DATA"
        ,"ENEMY_DEF_NAME", "QUEST_22_混成栽培兵器６号"
        ,"MAX_HP", 12000
        ,"STR", 120
        ,"MAG", 30
        ,"ATTACK", 100
        ,"DEFENCE", 30
        ,"ARMOR", 250
        ,"AGILITY", 60
        ,"HATE_RATE", 100
        ,"TARGETING", WPN_脅威選定_無作為
        ,"ENEMY_PASSIVE_SKILL", "[STUN_ENEMY:3]"
        ,"ENEMY_ACTIVE_SKILL", "[SMITE:2][ACT_TORN_CLOTH:1][ACT_FRONT_DEFENCE:3]"
        ,"DROP_ITEM_LIST", @"MONEY,2000,1000,{ITEM_魔獣の肉片},5,1000,{ITEM_植物性繊維},5,1000,"
        ,"EXP", 8
}
{
    DT_ROW_ADD "ENEMY_DATA"
        ,"ENEMY_DEF_NAME", "QUEST_22_茨の蛇蔦"
        ,"MAX_HP", 5000
        ,"STR", 80
        ,"MAG", 0
        ,"ATTACK", 80
        ,"DEFENCE", 15
        ,"ARMOR", 50
        ,"AGILITY", 120
        ,"HATE_RATE", 100
        ,"TARGETING", WPN_脅威選定_無作為
        ,"ENEMY_PASSIVE_SKILL", "[STUN_ENEMY:2]"
        ,"ENEMY_ACTIVE_SKILL", "[SMITE:1][ACT_TORN_CLOTH:1]"
        ,"DROP_ITEM_LIST", @"MONEY,500,1000,{ITEM_魔獣の肉片},1,1000,{ITEM_植物性繊維},1,1000,"
        ,"EXP", 2
}

path: ERB/QUEST/クエストデータ/QUEST_22_ノアレルム辺境の賊徒/QUEST_22_ノアレルム辺境の賊徒.ERB


code:
MAP_NAME = "ノアレルム辺境の賊徒"
MAP_IMAGE = "q22.webp"
PATH_COLOR = "0xf0ff"
PATH_ALPHA = "0x7d"
NODE_BASE_COLOR = "0xfff074"

path: ERB/QUEST/クエストデータ/QUEST_22_ノアレルム辺境の賊徒/ノアレルム辺境の賊徒.txt


code:
<DocumentElement>
  <GMAPDATA>
    <id>638533077702170603</id>
    <NODE_ID>0</NODE_ID>
    <NODE_NAME>入口</NODE_NAME>
    <POS_X>420</POS_X>
    <POS_Y>380</POS_Y>
    <PATH_LIST>1,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638533078211089875</id>
    <NODE_ID>1</NODE_ID>
    <NODE_NAME>木こりの掘立小屋</NODE_NAME>
    <POS_X>360</POS_X>
    <POS_Y>340</POS_Y>
    <PATH_LIST>0,2,3,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638533079815650352</id>
    <NODE_ID>2</NODE_ID>
    <NODE_NAME>小さな祠</NODE_NAME>
    <POS_X>460</POS_X>
    <POS_Y>280</POS_Y>
    <PATH_LIST>1,4,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638533080111569537</id>
    <NODE_ID>3</NODE_ID>
    <NODE_NAME>森の入り口</NODE_NAME>
    <POS_X>260</POS_X>
    <POS_Y>320</POS_Y>
    <PATH_LIST>1,5,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638533080686210046</id>
    <NODE_ID>4</NODE_ID>
    <NODE_NAME>荒廃した旧村道</NODE_NAME>
    <POS_X>500</POS_X>
    <POS_Y>260</POS_Y>
    <PATH_LIST>2,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638533082113410722</id>
    <NODE_ID>5</NODE_ID>
    <NODE_NAME>林間の踏み跡</NODE_NAME>
    <POS_X>160</POS_X>
    <POS_Y>300</POS_Y>
    <PATH_LIST>3,6,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638533083010049937</id>
    <NODE_ID>6</NODE_ID>
    <NODE_NAME>ガレた急斜面</NODE_NAME>
    <POS_X>100</POS_X>
    <POS_Y>220</POS_Y>
    <PATH_LIST>5,7,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638533083470049441</id>
    <NODE_ID>7</NODE_ID>
    <NODE_NAME>廃集落のはずれ</NODE_NAME>
    <POS_X>120</POS_X>
    <POS_Y>160</POS_Y>
    <PATH_LIST>6,8,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638533084939810579</id>
    <NODE_ID>8</NODE_ID>
    <NODE_NAME>廃屋の陰</NODE_NAME>
    <POS_X>200</POS_X>
    <POS_Y>160</POS_Y>
    <PATH_LIST>7,10,9,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638533085011089505</id>
    <NODE_ID>9</NODE_ID>
    <NODE_NAME>崩れかけた小屋</NODE_NAME>
    <POS_X>300</POS_X>
    <POS_Y>140</POS_Y>
    <PATH_LIST>8,12,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638533085106851219</id>
    <NODE_ID>10</NODE_ID>
    <NODE_NAME>古びた石造りの小屋</NODE_NAME>
    <POS_X>280</POS_X>
    <POS_Y>180</POS_Y>
    <PATH_LIST>8,11,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638533086252371571</id>
    <NODE_ID>11</NODE_ID>
    <NODE_NAME>隠された地下室</NODE_NAME>
    <POS_X>280</POS_X>
    <POS_Y>200</POS_Y>
    <PATH_LIST>10,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638533087910288713</id>
    <NODE_ID>12</NODE_ID>
    <NODE_NAME>大きな納屋</NODE_NAME>
    <POS_X>240</POS_X>
    <POS_Y>120</POS_Y>
    <PATH_LIST>9,</PATH_LIST>
  </GMAPDATA>
</DocumentElement>

path: ERB/QUEST/クエストデータ/QUEST_22_ノアレルム辺境の賊徒/ノアレルム辺境の賊徒.xml


code:
;-------------------------------------------------
;クエスト定義ファイル
;QUEST_23:開拓村の魔物退治
;-------------------------------------------------
@QST_TITLE_QUEST_23
RESULTS '= "飛竜退治"

;ギルドメニューでクエストをフィルターする際のタグリスト。末尾の','を忘れないように
@QST_QUEST_TAG_23
RESULTS '= "単発,探索,"

;クエストメニュー右上段の概要欄。10行まで
@QST_SUMALLY1_QUEST_23
RESULTS:0 '= "ノアレルム西方領の首長たちからの依頼だ"
RESULTS:1 '= "小型の飛竜……ワイバーンたちがねぐらを構えて周辺の集落に「ちょっかい」をかけているらしい"
RESULTS:2 '= "ワイバーンの巣の探索と駆除がお望みとのことだ"
RESULTS:4 '= "一頭一頭はさほど手強い相手ではないが、頭数は多そうだ"
RESULTS:5 '= "十分注意するに越したことはないだろう"

;クエストメニュー右下段の備考欄。難易度や条件、報酬等を記述。7行まで
@QST_SUMALLY2_QUEST_23
RESULTS:0 '= @"難易度：@C:{カラー_注意}@★×９@/C@"
RESULTS:1 '= "報酬：20,000ディナリ"
RESULTS:2 '= "推奨パーティ人数:3人以上　　　推奨平均戦闘値：90以上"
RESULTS:4 '= @"@C:{カラー_選択不可}@「ああ、あの辺は、昔は有名な長命竜の縄張りだった土地だな」@/C@"
RESULTS:5 '= @"@C:{カラー_選択不可}@「数年前に討伐されてからは、ごく平穏だったと聞いてるが……」@/C@"
RESULTS:6 '= @"@C:{カラー_選択不可}@「ノアレルム本国はリナシメントにご執心で、辺境警備まで気が回ってないんだろう」@/C@"

;-------------------------------------------------
;クエストの実行可能条件を定義
;-------------------------------------------------
@QST_AVAILABLE_QUEST_23(QUEST_ID)
#DIM QUEST_ID
;リナシメントの進捗スコア記録6以上で解禁
SIF QST_SV_QUEST_FLAG:9 <= 5
	RETURN 0
;QUEST_6_湯泡郷クリアが前提
SIF QST_SV_QUEST_FLAG:6 != 2
	RETURN 0
SIF QST_SV_QUEST_FLAG:QUEST_ID != 2
	RETURN 1
RETURN 0

;-------------------------------------------------
;クエスト固有の開始条件
;[このクエストを開始する]押下時に呼ばれる
;-------------------------------------------------
@QST_CAN_START_QUEST_23(QUEST_ID)
#DIM QUEST_ID
#DIM DYNAMIC MEMBER
#DIM DYNAMIC CHARA_ID
#DIM DYNAMIC NUM
LOCALS '= ""

FOR LOCAL, 0, LIST_COUNT(_LIST_QST_PARTY)
	CHARA_ID = TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))
	NUM ++
NEXT
IF NUM < 3
	CALLF TAG_PRINT(@"@C:{カラー_注意}@パーティ人数が不足しています(必須：３名以上)@/C@", 1)
	WAIT
	RETURN 0
ELSE
	RETURN 1
ENDIF

;-------------------------------------------------
;マップデータ定義
;[ROOM_ID:]　…　配置するエリアの種類
;[ROOM_PATH:]　…　エリア間の接続情報(※)
;[EVENT_LIST:]　…　エリアに配置するイベントのID(※)
;(※)要素の末尾の','を忘れないように注意
;EmueraをDEBUGMODEで起動した場合、[領地の手入れ]画面から編集したエリア情報をテキストファイルにエクスポートできる
;出力txtをベースに編集すると少し楽かもしれない
;-------------------------------------------------
@QST_SET_MAPDATA_QUEST_23(MAP_ID)
#DIM MAP_ID
QST_MAPNAME '= "飛竜のねぐら周辺"
_DIC_HO_MAPDATA:MAP_ID:0  '= "[ROOM_ID:201][MODULE_SLOT:][ROOM_PATH:1,][ROOM_NAME:洞窟の最奥][EVENT_LIST:0,130,]"
_DIC_HO_MAPDATA:MAP_ID:1  '= "[ROOM_ID:201][MODULE_SLOT:][ROOM_PATH:2,0,][ROOM_NAME:血濡れた隘路][EVENT_LIST:0,120,]"
_DIC_HO_MAPDATA:MAP_ID:2  '= "[ROOM_ID:201][MODULE_SLOT:][ROOM_PATH:1,7,][ROOM_NAME:広大な鍾乳洞][EVENT_LIST:0,110,]"
_DIC_HO_MAPDATA:MAP_ID:3  '= "[ROOM_ID:0]"
_DIC_HO_MAPDATA:MAP_ID:4  '= "[ROOM_ID:0]"
_DIC_HO_MAPDATA:MAP_ID:5  '= "[ROOM_ID:0]"
_DIC_HO_MAPDATA:MAP_ID:6  '= "[ROOM_ID:201][MODULE_SLOT:][ROOM_PATH:7,][ROOM_NAME:岩間の湧水][EVENT_LIST:0,100,]"
_DIC_HO_MAPDATA:MAP_ID:7  '= "[ROOM_ID:201][MODULE_SLOT:][ROOM_PATH:6,8,12,2,][ROOM_NAME:ねぐらの入口][EVENT_LIST:0,70,]"
_DIC_HO_MAPDATA:MAP_ID:8  '= "[ROOM_ID:201][MODULE_SLOT:][ROOM_PATH:7,9,][ROOM_NAME:抉られた風穴][EVENT_LIST:0,80,]"
_DIC_HO_MAPDATA:MAP_ID:9  '= "[ROOM_ID:201][MODULE_SLOT:][ROOM_PATH:8,][ROOM_NAME:貯食の洞穴][EVENT_LIST:0,90,]"
_DIC_HO_MAPDATA:MAP_ID:10  '= "[ROOM_ID:201][MODULE_SLOT:][ROOM_PATH:11,][ROOM_NAME:岩山の断崖][EVENT_LIST:0,60,]"
_DIC_HO_MAPDATA:MAP_ID:11  '= "[ROOM_ID:201][MODULE_SLOT:][ROOM_PATH:12,10,][ROOM_NAME:岩壁の獣道][EVENT_LIST:0,50,]"
_DIC_HO_MAPDATA:MAP_ID:12  '= "[ROOM_ID:201][MODULE_SLOT:][ROOM_PATH:7,11,17,][ROOM_NAME:山道の終端][EVENT_LIST:0,40,]"
_DIC_HO_MAPDATA:MAP_ID:13  '= "[ROOM_ID:0][MODULE_SLOT:][ROOM_PATH:]"
_DIC_HO_MAPDATA:MAP_ID:14  '= "[ROOM_ID:0][MODULE_SLOT:][ROOM_PATH:]"
_DIC_HO_MAPDATA:MAP_ID:15  '= "[ROOM_ID:0][MODULE_SLOT:][ROOM_PATH:]"
_DIC_HO_MAPDATA:MAP_ID:16  '= "[ROOM_ID:0][MODULE_SLOT:][ROOM_PATH:]"
_DIC_HO_MAPDATA:MAP_ID:17  '= "[ROOM_ID:201][MODULE_SLOT:][ROOM_PATH:12,18,][ROOM_NAME:小高い丘][EVENT_LIST:0,30,]"
_DIC_HO_MAPDATA:MAP_ID:18  '= "[ROOM_ID:201][MODULE_SLOT:][ROOM_PATH:17,19,][ROOM_NAME:林間道][EVENT_LIST:0,20,]"
_DIC_HO_MAPDATA:MAP_ID:19  '= "[ROOM_ID:201][MODULE_SLOT:][ROOM_PATH:18,24,][ROOM_NAME:風化した石碑][EVENT_LIST:0,10,]"
_DIC_HO_MAPDATA:MAP_ID:20  '= "[ROOM_ID:0][MODULE_SLOT:][ROOM_PATH:]"
_DIC_HO_MAPDATA:MAP_ID:21  '= "[ROOM_ID:0][MODULE_SLOT:][ROOM_PATH:]"
_DIC_HO_MAPDATA:MAP_ID:22  '= "[ROOM_ID:0][MODULE_SLOT:][ROOM_PATH:]"
_DIC_HO_MAPDATA:MAP_ID:23  '= "[ROOM_ID:0][MODULE_SLOT:][ROOM_PATH:]"
_DIC_HO_MAPDATA:MAP_ID:24  '= "[ROOM_ID:200][MODULE_SLOT:][ROOM_PATH:19,][ROOM_NAME:牧羊の村][EVENT_LIST:0,1,2,]"

;-------------------------------------------------
;クエストのタイプ
;マップ探索を伴う場合 "MAP"
;文章のみで完結する場合 "DIALOGUE"
;-------------------------------------------------
@QST_TYPE_QUEST_23
RESULTS '= "MAP"

;-------------------------------------------------
;クエスト固有の初期化処理
;-------------------------------------------------
@QST_INIT_QUEST_23
;クエスト固有のユニークエネミーをエネミー定義データに追加
CALL QST_QUEST_23_ENEMY_DATA

;-------------------------------------------------
; クエスト中止時の処理
; QST_QUEST_{QUEST_ID}_ABORT は撤退コマンド実行時に呼ばれる
;-------------------------------------------------
@QST_QUEST_22_ABORT

;-------------------------------------------------
;QST_INTRO_QUEST_n はQUEST開始前に呼ばれる
;クエストの導入文などを記述
;"DIALOGUE"型のクエストは当該関数のCALLのみ行う
;-------------------------------------------------
@QST_INTRO_QUEST_23
PRINTL
PRINTFORMW ……………………
PRINTFORMW ………………
PRINTFORMW …………
PRINTL
CALL PRINT_TITLE("飛竜退治の依頼", "QUEST")
PRINTL
IF !QST_SV_QUEST_FLAG:QST_有効クエストID
	PRINTFORMW %QST_LEADER_NAME()%が依頼主から指示された村落を訪れると、広大な牧草地が広がっていたが……
	PRINTFORMW そこにいるべき羊や牛などの家畜が、ほとんど見当たらない
	PRINTL
	PRINTFORML 「……ああ、冒険者の方々ですか。ようこそ、大したもてなしもできませんで……」
	PRINTL
	PRINTFORMW 憔悴しきっている感じがひしひしする、村長と思しき初老の男性が対応してくれた
	PRINTL
	PRINTFORMW 話を聞いてみると、ここ数か月でワイバーンの襲撃が頻発し、家畜という家畜が攫われてしまったという
	PRINTL
	PRINTFORML ノアレルムの軍が調査に入ったのは一度だけで、その時の山狩りでは成果が上がらず、
	PRINTFORMW その後この村も被害にあってしまったという
	PRINTL
	PRINTFORMW ワイバーンがやってくるのは、決まって集落の北方、石灰岩の岩峰地帯からであるという
	PRINTFORMW その辺りは鍾乳洞や洞穴が無数に穴をあけているらしく、ねぐらには事欠かないだろう、とのことだ
	PRINTL
	PRINTFORMW ここを拠点に準備を整え、北に向かうとしよう……
ELSE
	PRINTFORMW %QST_LEADER_NAME()%は魔物退治を再開すべく、再び牧羊の村落へと向かった……
ENDIF

PRINTL
CALL FL_PLAYBGM("Forest_2.ogg", 50)

;このフラグが立っている場合ROOM_ID200(開始地点)で撤退コマンドが実行可能になる
FLAG:クエスト_中断を許可 = 1
QST_SV_QUEST_FLAG:QST_有効クエストID = 1

;-------------------------------------------------
;イベントリストに登録するイベント情報を記述する
;<記述方法>
;EVENT_TRIGGER
; "ENTER_ROOM" イベントが配置されたルームの入室時に強制的に実行される
; "EVENT_SLOT" イベントリストのクリック時に実行される
; "TRAIN_EXIT" 調教イベント終了後 ※予め変数:QST_TRAIN_EXIT_EVENT_ID に調教終了後に実行したいイベントIDを指定する必要がある
;EVENT_SLOT_NAME
; イベントリスト上の表示名
;EVENT_TYPE
; "ONCE" クエスト中一度のみ呼出されるイベント(※トリガーが"ENTER_ROOM"のイベントのみ機能)
;           実行済みかをフラグ管理するまでもないイベント(会話など)で使用する想定
;
;戻り値
; RESULT:0 = EVENT_TRIGGER
; RESULT:1 = EVENT_SLOT_NAME
; RESULT:2 = EVENT_TYPE
;-------------------------------------------------
;-------------------------------------------------
;EV0
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_23_EVENT_0
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

;-------------------------------------------------
;実行するイベントの実行内容を記述
;イベント用の関数はTRYCALLFORMで呼ばれるため
;@QST_QUEST_{QUEST_ID}_EVENT_{EVENT_ID}の書式で記述する
;-------------------------------------------------
@QST_QUEST_23_EVENT_0
#DIMS ROOM_NAME

CALL HO_CREAR_LOG()
SELECTCASE HO_MASTER現在地
	CASE 24;牧羊の村・スタート地点
		CALL HO_PRINT_LOG("牧羊の村は静まり返っている……", 1)
	CASE 19;風化した石碑
		CALL HO_PRINT_LOG("風化した石碑が藪に埋もれるように立っている……存外に立派なものだ", 1)
	CASE 18;林間道
		CALL HO_PRINT_LOG("ゆるやかな林間の道が、丘陵まで続いている……", 1)
	CASE 17;小高い丘
		CALL HO_PRINT_LOG("見晴らしのいい丘陵に出た", 1)
		CALL HO_PRINT_LOG("目的の石灰岩峰群が北方に突き出しているのが見える……", 1)
		CALL HO_PRINT_LOG("周囲には、祭祀場の跡と思しき石碑や石舞台の痕跡が見て取れる", 1)
	CASE 12;山道の終端
		CALL HO_PRINT_LOG("辛うじてついていた道が、ここで途切れている", 1)
		CALL HO_PRINT_LOG("いつ魔物が飛び出してきてもおかしくない雰囲気だ……", 1)
	CASE 11;岩壁の獣道
		CALL HO_PRINT_LOG("そり立つ岩壁をへつるように、か細い踏み跡が見える", 1)
		CALL HO_PRINT_LOG("……微かにだが、狼らしき遠吠えが聞こえる", 1)
	CASE 10;岩山の断崖
		CALL HO_PRINT_LOG("先ほどまでダイアウルフがいた場所だ", 1)
		CALL HO_PRINT_LOG("……遠く、狼の遠吠えが聞こえた気がする", 1)
	CASE 7;ねぐらの入口
		CALL HO_PRINT_LOG("ひときわ大きな洞窟が口を開けている", 1)
		CALL HO_PRINT_LOG("飛竜の足跡の痕跡も見受けられる……ここが、奴らのねぐらだろうか", 1)
	CASE 6;岩間の湧水
		CALL HO_PRINT_LOG("岩盤の隙間から、滾々と水が湧き出している……", 1)
	CASE 8;抉られた風穴
		CALL HO_PRINT_LOG("洞窟の中にワイバーンの足跡が続いている……", 1)
		CALL HO_PRINT_LOG("微かな腐臭が鼻を突いた……", 1)
	CASE 9;貯食の洞穴
		CALL HO_PRINT_LOG("洞窟の奥に、羊や牛のものと思われる骨屑や食い残しの腐肉がうず高く積み上げられている", 1)
		CALL HO_PRINT_LOG("ひどい腐臭がする……", 1)
	CASE 2;広大な鍾乳洞
		CALL HO_PRINT_LOG("鍾乳洞の広大な空間が広がっている……", 1)
		CALL HO_PRINT_LOG("微かな水の流れる音がどこからか聞こえる……", 1)
	CASE 1;血濡れた隘路
		CALL HO_PRINT_LOG("岩壁一面に血がぶちまけられている", 1)
		CALL HO_PRINT_LOG("道をふさぐように、ワイバーンの骸が転がっている", 1)
		CALL HO_PRINT_LOG("焼け焦げ、首を捩じ切られ、四肢がもぎ取られ……どれもひどく損壊している", 1)
	CASEELSE
		CALL HO_CREAR_LOG()
ENDSELECT

;-------------------------------------------------
;MAPID_24: EV1～
;牧羊の村
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_23_EVENT_1
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "村の長老"

RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
RESULTS:3 '= "[[EVENT_ICON:TALK]]"

@QST_QUEST_23_EVENT_1

PRINTL
PRINTFORMW 「羊たちを奪われては、もはや我々の集落の命脈も尽きたも同然じゃ」
PRINTL
PRINTFORMW 「こんな状況に陥ってから、ずっと考えていたんじゃよ」
PRINTFORMW 「これが……大古竜の縄張りを奪い取った我らへの天罰なのやもしれん、とな」
PRINTL
PRINTFORMW 「……いかんの、歳を取ると悲観的になってかなわん。忘れとくれ」
PRINTW

@QST_ADD_EVENT_QUEST_23_EVENT_2
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "羊飼いの子ども"

RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
RESULTS:3 '= "[[EVENT_ICON:TALK]]"

@QST_QUEST_23_EVENT_2

PRINTL
PRINTFORMW 「みんな信じてくれないけど……この目で、確かに見たんだ」
PRINTL
PRINTFORMW 「山道を、小さな女の子が飛ぶような足取りで駆けて行ったんだよ」
PRINTFORMW 「魔物が出るような山に分け入ってくわけがないって、僕もそう思うけど……」
PRINTL
PRINTFORMW 「冒険者さんたち……もしも、があったら、よろしく頼むよ」
PRINTW

;-------------------------------------------------
;MAPID_19:EV10～
;風化した石碑
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_23_EVENT_10
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "風化した石碑"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
RESULTS:3 '= "[EVENT_ICON:CHESS-ROOK]"

@QST_QUEST_23_EVENT_10

PRINTL
PRINTFORMW %QST_LEADER_NAME(1)%は、半ば藪に覆い隠されるように鎮座した石碑に目を止めた
PRINTFORMW 文字は所々欠けているが、おおよそこんなことが書いてあるようだ
PRINTL
PRINTFORMW ――大古竜ファスティは当地の守り手　悠久の時を超える語り部である
PRINTFORMW 彼女は人に作付けの時を教え、大嵐を教え、実りを教え、大雪を教えた
PRINTL
PRINTFORMW ただし、彼女は人を守るのではなく、自分の領域を守っているだけだ
PRINTL
PRINTFORMW ……ゆめ忘れることなかれ
PRINTL
PRINTFORMW 我らは彼女のひざ元で、彼女に奉仕し、その慈悲の下で生きているのだと――
PRINTL
PRINTFORMW …………
PRINTFORMW ………
PRINTFORMW ……
PRINTL
PRINTFORMW ……この地にかつて在ったという大古竜のことを記した碑文であろうか
PRINTFORMW 太古より、人とはある程度適切な距離感を保って暮らしてきたようだ
PRINTL
PRINTFORMW 数年前に討伐が行われたとのことだが、どんな経緯だったのだろうか……？
PRINTW
CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)

;-------------------------------------------------
;MAPID_18:EV20～
;ゆるやかな林間道
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_23_EVENT_20
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_23_EVENT_20
#DIM EVENT_FLAG
#DIM TARGET_ID

PRINTFORMW %QST_LEADER_NAME()%は林の中、ゆるやかに傾斜を強めていく山道を進んでいく
PRINTFORMW ……遠く、狼の遠吠えが聞こえた気がした
PRINTL
PRINTFORMW まだこの辺りは、木こりたちも立ち入る領域だからか、道もしっかりとついている
PRINTL
PRINTFORMW とはいえ、相手はワイバーンだ
PRINTFORMW 梢の隙間より見える上空への警戒も怠るべきではないだろう……
PRINTW

;-------------------------------------------------
;MAPID_17:EV30～
;小高い丘
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_23_EVENT_30
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

;ゲーム中一度でも解決済みの場合は非アクティブに
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT
	CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ENDIF

@QST_QUEST_23_EVENT_30
#DIM EVENT_FLAG
#DIM TARGET_ID

PRINTFORMW %QST_LEADER_NAME()%は林を抜け、開けた場所に出た
PRINTFORMW どうやら上り坂もここまで。丘陵の頂上に立ったようだ
PRINTL
PRINTFORMW 丘の上は木々が刈り払われた痕跡が見受けられ、やや広い平坦地……広場が確保されている
PRINTL
PRINTFORML 山城の跡かとも思ったが、所々にある石碑や、周囲より一段高くなった石舞台と思しき遺構を見ると、
PRINTFORMW どうやら祭祀場でもあったかのような雰囲気が感じられた
PRINTL
PRINTFORML %QST_LEADER_NAME(1)%がしばし休憩でも……と考えていると
PRINTFORMW 突如聞こえた耳をつんざくような金切り音にびくり、とする
PRINTL
PRINTFORMW すわ、ワイバーンのおでましかと思えば……
PRINTFORMW 上空に飛来した生物の影に、目を見張った
PRINTL
PRINTFORMW それは――大きな目玉に、コウモリの翼とタコの触手をくっつけたような奇怪なフォルム
PRINTFORMW 数体の怪生物が、こちらに急降下攻撃をかけてくる！
PRINTL

;戦闘イベント
CALL CC_INIT()
;クエストのパーティメンバー全員を攻撃側として戦闘メンバーに追加
CALL CC_ADDPARTY()
;エネミーを防御側で追加
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_23_ウォッチャー", "ウォッチャー１")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_23_ウォッチャー", "ウォッチャー２")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_23_ウォッチャー", "ウォッチャー３")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_23_ウォッチャー", "ウォッチャー４")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_23_ウォッチャー", "ウォッチャー５")
CALL CLOSE_COMBAT("[BGM:Battle_1.ogg]")
IF RESULT == 1
	PRINTL
	;戦闘で増減したステータスをキャラに反映
	CALL CC_APPLY_CHARA_HP()
	;戦利品獲得処理
	CALL CC_GET_LOOT_FROM("DEFENDER")
	PRINTW
	PRINTFORMW %ANAME(QST_RANDCHARA())%が得物を振るって怪生物を追い散らす……
	PRINTFORMW %QST_LEADER_NAME()%は周囲を見渡し、再度の襲撃がないことを確認すると、一息ついた
	PRINTL
	PRINTFORMW 先ほどの怪生物は……言うほど、強い魔物というわけではなかった
	PRINTFORMW だが、妙に使い魔めいた外見とその挙動が、少し気にかかった
	PRINTL
	PRINTFORMW まるで、高所から何かを見張っていたかのような……
	PRINTL
	PRINTFORMW ……少しばかりうすら寒いものを感じながら、%QST_LEADER_NAME()%は行く手を見渡した
	PRINTFORMW 北方に、石灰岩の白い大岩峰がはっきりと視認できる
	PRINTFORMW 遠目には、ワイバーンの姿は見当たらない……もう少し、近付いてみる必要がありそうだ
	PRINTL
	CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)
ELSE
	CALL CC_APPLY_CHARA_HP()
	PRINTL
	PRINTFORMW 飛来する怪生物に翻弄されてしまい、傷ばかりが増えていく……
	PRINTFORMW このままでは勝ち目が薄い、戦い続けても命を落とすだけであろう
	PRINTL
	PRINTFORMW %QST_LEADER_NAME()%は包囲を破り、退却することを決意した……！
	;汎用敗北イベントへ
	CALL QST_QUEST_23_DEFEAT_EVENT
ENDIF

;-------------------------------------------------
;MAPID_12:EV40～
;山道の終点
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_23_EVENT_40
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_23_EVENT_40
#DIM EVENT_FLAG
#DIM TARGET_ID

PRINTFORMW %QST_LEADER_NAME()%は丘陵から九十九折りに下る山道を辿って、鞍部まで降りてきた
PRINTL
PRINTFORMW 辛うじて保たれていた山道が、ここで終点を迎えたようだ
PRINTFORMW 現在地は石灰岩峰群のちょうど真下
PRINTFORMW 見上げれば、巨岩奇岩がそこらじゅうの地面より突き出し、露出した白い岩壁が高くそり立っている
PRINTL
PRINTFORMW 目の前にはぽっかりと空いた巨大な洞窟の入口
PRINTFORMW 左には、岩壁をへつるように続く踏み跡が残されている
PRINTL
PRINTFORMW 狼の遠吠えが、頭上、遠方より聞こえた気もする
PRINTFORMW さて、どちらに進んだものだろうか……
PRINTW

;-------------------------------------------------
;MAPID_11:EV50～
;岩峰をへつる獣道
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_23_EVENT_50
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_23_EVENT_50
#DIM EVENT_FLAG
#DIM TARGET_ID

PRINTFORMW %QST_LEADER_NAME()%は岩峰をへつって、上へ上へと続くか細い踏み跡を進んでいく……
PRINTL
PRINTFORMW 所々、足場が不安定で壁に張り付くように動いたり、オーバーハングした岩壁を屈んで躱す必要もあった
PRINTFORMW どうもこの踏み跡は、人間がつけたものではないようだ……
PRINTL
PRINTFORMW ……ひょっとして、ここで襲われたらひとたまりもないのでは？
PRINTL
PRINTFORMW 小さく身震いしながら、さっさと通り抜けるに限る、とばかりに足を速める
PRINTL
PRINTFORMW どんどん心許なくなる足場に、引き返すタイミングを逸した感が否めない……
PRINTW

;-------------------------------------------------
;MAPID_10:EV60～
;岩山の断崖
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_23_EVENT_60
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

;ゲーム中一度でも解決済みの場合は非アクティブに
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT
	CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ENDIF

@QST_QUEST_23_EVENT_60
#DIM EVENT_FLAG
#DIM TARGET_ID

CALL PRINT_TITLE("大狼のお節介", "EVENT")
CALL DRAW_STILL(@"STILL_ダイアウルフ", @"[SIZE_X:480px][SIZE_Y:320px]", 1)

PRINTFORMW %QST_LEADER_NAME()%が苦労して岩峰を登りきると、その断崖には三頭の狼が佇んでいた
PRINTL
;深森の忘都で『ファヌイリエルの賢狼』に勝利している
IF QST_SV_EVENT_FLAG:30:131 == 1
	PRINTFORMW ……見紛うはずもない。ファヌイリエルの中腹やローレリアム跡で対峙したダイアウルフの群れだ
ELSE
	PRINTFORMW それはかつて、ファヌイリエルの中腹で一行が対峙したダイアウルフの群れであった
ENDIF
PRINTL
PRINTFORMW リラックスした様子の彼らからは、こちらに対する敵意は感じられない
PRINTFORMW なぜこんな所にいるのかはわからないが、彼らの力であれば遊弋するワイバーンなどさして問題にはならないであろう
PRINTL
PRINTFORMW ……ダイアウルフの長は、相変わらずこちらを見定めるかのような、力のある強い双眸を向けてくる
PRINTL
IF QST_SV_EVENT_FLAG:30:131 == 1
	PRINTFORMW ローレリアムの丘で繰り広げたあの戦いの再演を望んでいるのか――
	PRINTFORMW ……と思ったが、どうやらそのような雰囲気ではないらしい
	PRINTL
ENDIF
PRINTFORMW やがて、しばしの沈黙ののち、大狼は金色の瞳を細めてこちらを見やると……
PRINTFORMW ゆっくりと頭を振り、崖下に大きく口を開けた洞窟を示した
PRINTFORMW その仕草が、なぜか――
PRINTL
PRINTFORMW （――あとは任せた）
PRINTL
PRINTFORMW そう、こちらに告げているようで
PRINTL
PRINTFORMW ワイバーンを倒せ、ということか？　それとも、別の意味が？
PRINTL
PRINTFORMW 意味がわからず%QST_LEADER_NAME(1)%が困惑していると、大狼は愉快気に鼻を鳴らし――笑ったのか？
PRINTFORMW ダイアウルフたちを伴ってゆっくりと立ち上がり、その身を、断崖の下へと躍らせる
PRINTL
PRINTFORMW 巨体に似合わぬ、軽々とした足取り。岩壁のわずかな足掛かりを正確に辿り、あっという間に崖下へと降り立った
PRINTL
PRINTFORMW 「…………」
PRINTL
PRINTFORMW 遠く念押しのような金色の一瞥をくれて、大狼がダイアウルフの群れを率いて森に消えていく
PRINTL
PRINTFORMW 一体何だったのだろう……
PRINTFORMW 何か厄介事を背負わされたのではないか、と疑念に囚われる%QST_LEADER_NAME(1)%だった
CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)
PRINTW

;-------------------------------------------------
;MAPID_7:EV70～
;ねぐらの入口
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_23_EVENT_70
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_23_EVENT_70
#DIM EVENT_FLAG
#DIM TARGET_ID

PRINTFORMW %QST_LEADER_NAME()%は足音を潜めて大きな洞窟の入口にとりついた
PRINTFORMW 外気とは明確に異なる冷涼な空気が、地下から吹き上がってくるのを感じる……
PRINTL
PRINTFORMW ここに至るまで、ワイバーンのものとはっきりわかる足跡がいくつも見つかった
PRINTFORMW おそらくここが、連中のねぐらと見て間違いないだろう
PRINTL
PRINTFORMW 洞窟の奥、ぬばたまの闇の向こうはまるで見通せない
PRINTFORMW この鍾乳洞は相当に大規模だと聞いているが、この中ではワイバーンの飛翔能力も大幅に制限される
PRINTFORMW 始末をつけるなら洞窟の中で。一頭たりとも逃がさない――
PRINTL
PRINTFORMW %QST_LEADER_NAME()%は入念に装備を確認し、意を決して洞窟の中に足を踏み入れた……
PRINTW

;-------------------------------------------------
;MAPID_8:EV80～
;抉られた風穴
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_23_EVENT_80
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_23_EVENT_80
#DIM EVENT_FLAG
#DIM TARGET_ID

PRINTFORMW %QST_LEADER_NAME()%は、人間には十分余裕のある風穴を進んでいく
PRINTL
PRINTFORMW 所々に、何かが擦りつけられたような跡や、硬質なもので削り取られた跡がある
PRINTFORMW おそらく、図体の大きいワイバーンたちが行き交った跡なのだろう
PRINTL
PRINTFORMW 前方からは、鼻を突くようなすえた匂いが漂ってくる
PRINTFORMW 何が待ち受けているのか。少しばかり気を張りながら、慎重に歩みを進めていく……
PRINTW

;-------------------------------------------------
;MAPID_9:EV90～
;貯食の洞穴
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_23_EVENT_90
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

;ゲーム中一度でも解決済みの場合は非アクティブに
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT
	CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ENDIF

@QST_QUEST_23_EVENT_90
#DIM EVENT_FLAG
#DIM TARGET_ID

PRINTFORMW 風穴の最奥にたどり着いた%QST_LEADER_NAME(1)%は眉をひそめた
PRINTFORMW 先刻から漂っていた悪臭の正体が判明したからだ
PRINTL
PRINTFORMW おそらくは羊や牛のものと思われる骨片や、食べ残しの肉が腐ったもの……
PRINTFORMW それらが、ほら穴一杯にうず高く積み上げられている
PRINTL
PRINTFORMW これが、ワイバーンの貯食行動なのだろうか
PRINTFORMW しかし、骨についた肉や骨髄をただ腐らせるだけとは、あまりに無駄が多いような……
PRINTL
PRINTFORMW と、死骸の山を検分していた%ANAME(QST_RANDCHARA())%が、警戒の声を上げる
PRINTL
PRINTFORMW 促された先を見やると、骨片と腐肉の壁を掻き分けて、うぞうぞと何かが這い出てくる
PRINTFORMW 腐肉を漁るナメクジのような魔物が、死骸の山を鳴動させながら襲い掛かってきた！
PRINTL

;戦闘イベント
CALL CC_INIT()
;クエストのパーティメンバー全員を攻撃側として戦闘メンバーに追加
CALL CC_ADDPARTY()
;エネミーを防御側で追加
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "ヒュージ・スラッグ", "肥え太った肉食ナメクジ", 200)
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "ヒュージ・スラッグ", "肉食ナメクジ１")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "ヒュージ・スラッグ", "肉食ナメクジ２")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "ヒュージ・スラッグ", "肉食ナメクジ３")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "ヒュージ・スラッグ", "肉食ナメクジ４")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "ヒュージ・スラッグ", "肉食ナメクジ５")
CALL CLOSE_COMBAT("[BGM:Battle_1.ogg]")
IF RESULT == 1
	PRINTL
	;戦闘で増減したステータスをキャラに反映
	CALL CC_APPLY_CHARA_HP()
	;戦利品獲得処理なし
	PRINTW
	PRINTFORMW %ANAME(QST_RANDCHARA())%が得物を叩きつけ、体液をまきちらして最後の肉食ナメクジが絶命する
	PRINTFORMW %QST_LEADER_NAME()%は周囲を見渡し、生き残っている肉食ナメクジがいないことを確認した
	PRINTL
	PRINTFORMW ……こいつらは、おそらくワイバーンにとっての「家畜」のようなものだったのだろう
	PRINTFORMW 食べ残しの腐肉を与え、新鮮な肉を作り出せるのなら願ったりかなったりというわけだ
	PRINTL
	PRINTFORMW どうやら、ここはハズレのようだ……
	PRINTL
	PRINTFORMW 去ろうとした一行を、%ANAME(QST_RANDCHARA())%が呼び止める
	PRINTFORMW 飛竜が貯め込んだものか、キラキラと光を照り返す塊が、腐肉や骨片、体毛などの混合物の中に見えた
	PRINTL
	PRINTFORMW おそらくは、銀や金のインゴットであろうか
	PRINTL
	PRINTFORMW ……粘液に塗れてはいるが、まあ、貴金属なら大丈夫だろう、たぶん
	PRINTFORMW 行き掛けの駄賃として頂いていくことにした……
	PRINTL
	CALL QST_CHANGE_ITEM(@"{ITEM_銀インゴット},5,{ITEM_金インゴット},5,")
	CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)
ELSE
	CALL CC_APPLY_CHARA_HP()
	PRINTL
	PRINTFORMW 次から次へと現れるスラッグを相手にし続けるのもキリがない……
	PRINTFORMW このままでは勝ち目が薄い、戦い続けても命を落とすだけであろう
	PRINTL
	PRINTFORMW %QST_LEADER_NAME()%は包囲を破り、退却することを決意した……！
	;汎用敗北イベントへ
	CALL QST_QUEST_23_DEFEAT_EVENT
ENDIF

;-------------------------------------------------
;MAPID_6:EV100～
;岩間の湧水
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_23_EVENT_100
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "岩間の湧水"

RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
RESULTS:3 '= "[EVENT_ICON:CHESS-ROOK]"

@QST_QUEST_23_EVENT_100
#DIM EVENT_FLAG
#DIM TARGET_ID

PRINTFORML %ANAME(QST_RANDCHARA())%は岩間に滾々と湧き出す湧水を見つけた
PRINTFORMW 絶えず湧き出る水は清純で、喉の乾きを癒やすにはちょうど良いように見える
PRINTL
CALL ASK_YN("休憩する", "やめておく")
IF !RESULT
	PRINTFORMW %QST_LEADER_NAME()%は小川で喉の乾きを癒やした
	;パーティ全員の体力を回復
	FOR LOCAL, 0, QST_PARTYNUM
		TARGET_ID = TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))
		CALL QST_CHANGE_HP(TARGET_ID, 1000)
	NEXT
	WAIT
	CALL QST_DEACTIVATE_THIS_EVENT()
ELSE
ENDIF
CALL QST_SET_EVENT_FLAG(QST_NOWEV, 1)

;-------------------------------------------------
;MAPID_2:EV110～
;広大な鍾乳洞
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_23_EVENT_110
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

;ゲーム中一度でも解決済みの場合は非アクティブに
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT
	CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ENDIF

@QST_QUEST_23_EVENT_110
#DIM EVENT_FLAG
#DIM TARGET_ID

PRINTFORMW 鍾乳洞の核心部に至った%QST_LEADER_NAME()%は周囲に広がる光景に息を飲んだ
PRINTFORMW 高い天井から垂れ下がる無数の鍾乳石、地面から無数に伸びる石筍、それらが接合した堂々たる石柱
PRINTFORMW 流れる清水が形成した、棚田のようなリムストーンと、透き通った水を湛えたリムストーンプール……
PRINTL
PRINTFORMW ……こういう依頼でなければ、ゆっくり見物していきたいものだが、そうも言っていられない
PRINTFORMW 照明を翳しながら、ゆっくり慎重に洞窟の中を進んでいく……
PRINTL
PRINTFORMW 鍾乳洞の奥は地下河川が流れているらしく、どこからか流れる水の音が聞こえる
PRINTL
PRINTFORMW ……流れる水の音に紛れて、速いテンポの硬質な音が聞こえてくる
PRINTFORMW 注意深く照明を向けると――ワイバーンが数頭、こちらに向かって突進してきている！
PRINTL
PRINTFORMW うろたえず、%QST_LEADER_NAME()%は得物を翳して、迎撃態勢をとった――
PRINTL

;戦闘イベント
CALL CC_INIT()
;クエストのパーティメンバー全員を攻撃側として戦闘メンバーに追加
CALL CC_ADDPARTY()
;エネミーを防御側で追加
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "エルダーワイバーン", "狩り立てられたワイバーン１")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "エルダーワイバーン", "狩り立てられたワイバーン２")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "ワイバーン", "恐れ慄くワイバーン１")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "ワイバーン", "恐れ慄くワイバーン２")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "ワイバーン", "恐れ慄くワイバーン３")
CALL CLOSE_COMBAT("[BGM:Battle_1.ogg]")
IF RESULT == 1
	PRINTL
	;戦闘で増減したステータスをキャラに反映
	CALL CC_APPLY_CHARA_HP()
	;戦利品獲得処理なし
	PRINTW
	PRINTFORMW %ANAME(QST_RANDCHARA())%が逃げ延びようとするワイバーンの背に飛び乗り、頚椎に致命の一撃を喰らわした
	PRINTFORMW ずん、と洞窟内にワイバーンのくずおれる音が響き……再び、水の音だけが響き渡る
	PRINTL
	PRINTFORMW 一息ついて、周囲に倒れ伏したワイバーンたちを一頭一頭検分する
	PRINTFORMW ここまでで、五頭のワイバーンの群れと遭遇し、これを討ち果たしたわけだが……
	PRINTFORMW どうも、事が終わった気がしない
	PRINTL
	PRINTFORMW ……このワイバーンたちが、自分たちとは別のものに追い立てられて、必死に逃げてきたようにも見受けられたからだ
	PRINTFORMW なおも闇深く、洞窟は続いている
	PRINTFORMW この先も、注意を怠るべきではないだろう……
	PRINTL
	CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)
ELSE
	CALL CC_APPLY_CHARA_HP()
	PRINTL
	PRINTFORMW 鬼気迫る迫力のワイバーンたちの突進の勢いに負け、劣勢に追い込まれてしまう
	PRINTFORMW このままでは勝ち目が薄い、戦い続けても命を落とすだけであろう
	PRINTL
	PRINTFORMW %QST_LEADER_NAME()%は見切りをつけ、仕切り直すことを決意した……！
	;汎用敗北イベントへ
	CALL QST_QUEST_23_DEFEAT_EVENT
ENDIF

;-------------------------------------------------
;MAPID_1:EV120～
;血濡れた隘路
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_23_EVENT_120
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_23_EVENT_120
#DIM EVENT_FLAG
#DIM TARGET_ID

CALL FL_STOPBGM()
PRINTFORMW %QST_LEADER_NAME(1)%は闇の中、違和感に足を止めた
PRINTFORMW ……洞窟の冷たい風に乗って、微かに焦げ臭い匂いが漂ってきている、気がする
PRINTL
PRINTFORMW 同時、%ANAME(QST_RANDCHARA())%が、眼下の闇に向けて何の気なしに照明を向けると――
PRINTL
CALL FL_PLAYBGM("Ambient_dark_2.mp3")
PRINTFORMW ぎょっ、とした
PRINTFORMW ……焼け焦げたワイバーンの頭部……生首が、無造作に転がっていたからだ
PRINTL
PRINTFORMW そのままぐるり、と照明を巡らす
PRINTFORMW 白いはずの壁面……フローストーンはぶちまけられた鮮血で赤く染まり、水たまりは血で黒々と濁っている
PRINTFORMW 二体、三体……ワイバーンの無残な遺骸が、隘路を塞ぐように地に伏していた
PRINTL
PRINTFORMW ある死体は身体を二つ折りにされ、またある死体は四肢を無くし、ある死体は頭部を失っている
PRINTL
PRINTFORMW 死体を検分した%QST_LEADER_NAME(1)%は、その異様さに少し身震いした
PRINTFORMW どう見ても、「力任せ」に「捩じ切られた」ようにしか見えなかったからだ
PRINTL
PRINTFORMW ……この先に、非常識なパワーを持った何かがいることは、確かなようだ
PRINTFORMW しっかりと準備をして、踏み込むべきであろう
PRINTW

;-------------------------------------------------
;MAPID_0:EV130～
;洞窟の最奥
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_23_EVENT_130
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_23_EVENT_130
#DIM EVENT_FLAG
#DIM DYNAMIC MEMBER
#DIM DYNAMIC CHARA_ID
#DIM DYNAMIC MAGIC_POWER

;パーティ内で最も魔力の値が高い者をセレクト　フレーバーとして
FOR LOCAL, 0, LIST_COUNT(_LIST_QST_PARTY)
	CHARA_ID = TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))
	IF MAGIC_POWER < ABL:(CHARA_ID):魔術
		MAGIC_POWER = ABL:(CHARA_ID):魔術
		MEMBER = CHARA_ID
	ENDIF
NEXT

CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT != 1
	PRINTFORMW 洞窟に響き渡る悲鳴のような、ワイバーンの金切り声に、%QST_LEADER_NAME()%は思わず身構える
	PRINTFORMW ……そっと岩越しに見た光景に、思わず目を疑った
	PRINTL
	PRINTFORMW 子どものような小柄な人影が、小さいとはいえ竜の眷属のワイバーンの喉首を、片手で抑え込んでいる――
	PRINTL
	PRINTFORMW ジタバタともがくワイバーンにぴくりともせず、人影が大きく息を吸い込んだ、その刹那
	PRINTL
	CALL FL_PLAYSE("Explode_0.mp3")
	PRINTFORMW 闇を祓う閃光と衝撃が、洞窟を物理的に揺らした
	TWAIT 500, 1
	PRINTL
	PRINTFORMW パチリ、と弾けた雷光　焼け焦げる肉の匂い
	CALL FL_PLAYSE("HeavySmash_1.mp3")
	PRINTFORMW 丸焼けにされた「ワイバーンだったもの」が、ドサリ、と骸を晒す
	PRINTL
	PRINTFORMW たったの一撃で……と驚く間もなく、子どもの形をした何者かが、闇の中からこちらを睥睨する
	PRINTL
	PRINTFORMW ……ぞっ、と全身が総毛だった
ELSE
	PRINTFORMW 洞窟の最奥では、眠るように、子どものような姿をした何者かが大岩にもたれかかっていた
	PRINTFORMW まるでこちらの来訪を察知したかのように、「ソレ」がゆっくりと体を起こす……
ENDIF
CALL FL_PLAYBGM("Battle_3.ogg")
PRINTL
PRINTFORMW 頭部に突き出た角、背後に長く伸びた尻尾……いずれもヒトにはあり得ぬ異形
PRINTL
PRINTFORMW 闇の中に浮かび上がる双眸、その視線の強さ冷たさは、もはや人域のソレではない
PRINTFORMW その眼差しだけで、問答無用、話は通じないことが確信できた
PRINTL
PRINTFORMW 子どもの姿をした何者かが、すう、と大きく息を吸い込み――
CALL FL_PLAYSE("Wind_0.mp3")
PRINTFORMW 轟、と吹き抜けた一陣の風
PRINTFORMW 刹那に転輪する魔力が物理的な干渉力を帯び、大気に爆ぜて燐光を散らす
PRINTL
PRINTFORMW 「…………ッ！！」
PRINTFORMW %ANAME(MEMBER)%が、大仰な程にぶるりと身を震わせる
PRINTL
PRINTFORMW いかに魔術に疎くても、この巨きな『力』はわかる、ひしひしと伝わってくる
PRINTFORMW 人間としての本能が、畏れが、脳内で警報をかき鳴らし続けている
PRINTL
PRINTFORMW これは……『竜』
PRINTFORMW まさしく、真正の竜が汲み出す、マナの迸りだ！
PRINTL
PRINTFORMW 獣じみた唸り声。こちらへの鋭い敵意を剥き出しにした小柄な人影が、ゆらり、と右の腕を振り翳し――
CALL FL_PLAYSE("Fire_0.mp3")
PRINTFORMW 刹那に収束した魔力が、半透明な巨腕……雷霆を纏う「竜の腕」を具現化させる
PRINTL
PRINTFORMW ばちり、と弾けた稲光が、闇のヴェールに包まれたその顔を、ひと時だけ照らし出す
PRINTFORMW %QST_LEADER_NAME(1)%は、思わず息を呑んだ
PRINTFORMW まだ年端も行かぬ少女の面差しを、そこに見た気がして
PRINTL
PRINTFORMW ……何者かはわからないが、ここでやられるわけにはいかない
PRINTFORMW 気圧されそうになりながらも、人型の竜と思しき存在に相対し、%QST_LEADER_NAME()%は武器を執った――
;戦闘イベント
CALL CC_INIT()
;クエストのパーティメンバー全員を攻撃側として戦闘メンバーに追加
CALL CC_ADDPARTY()
;エネミーを防御側で追加
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_23_謎の少女", "人型の竜？")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_23_ダイアウルフの幻影", "ダイアウルフの幻影１")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_23_ダイアウルフの幻影", "ダイアウルフの幻影２")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_23_ダイアウルフの幻影", "ダイアウルフの幻影３")
CALL CLOSE_COMBAT("[BGM_NOT_AUTOSTOP:1]")
IF RESULT == 1
	PRINTL
	PRINTFORMW 「…………！！」
	PRINTL
	CALL FL_PLAYSE("Avoid_0.mp3")
	PRINTFORMW 蜻蛉を切って着地した人型の竜が、じり、と後ずさる
	PRINTL
	PRINTFORMW さすがに不利を悟ったか――
	PRINTFORML と思いきや、激しい戦闘の中で立ち位置が完全に入れ替わり、
	PRINTFORMW %QST_LEADER_NAME()%が洞窟の奥に押しやられる形になっている……！
	PRINTL
	PRINTFORMW 「…………」
	PRINTL
	PRINTFORMW つい、とこちらから視線を切った人型の竜は、その竜腕を勢いよく地面に叩きつける
	CALL FL_PLAYSE("Explode_2.mp3")
	PRINTFORMW 炸裂する雷光、遅れた衝撃に洞窟が激震
	TWAIT 750, 1
	CALL FL_PLAYSE("Explode_2.mp3")
	PRINTFORMW 壁面に大きな亀裂が走り、鍾乳石が大小を問わずばらばらと降り注ぐ――！
	TWAIT 750, 1
	CALL FL_PLAYSE("Explode_2.mp3")
	PRINTL
	PRINTFORMW ……これでは戦いどころではない！
	PRINTFORMW %QST_LEADER_NAME()%は落石から必死に身を守る――
	CALL FL_STOPBGM()
	PRINTL
	PRINTFORMW …………
	PRINTFORMW ………
	PRINTFORMW ……
	PRINTFORMW …
	PRINTL
	CALL FL_PLAYBGM("Ambient_dark_6.mp3")
	PRINTFORMW ……降り注いだ岩の隙間から、這う這うの体で這い出して、周囲を伺う
	PRINTFORMW 隙に乗じて、人型の竜は、完全に姿を晦ましたようだ
	PRINTL
	PRINTFORMW ……ここに留まるのも少々危険だ
	PRINTFORMW 洞窟が崩落する前に、早々に立ち去った方がよいだろう
	PRINTL
	PRINTFORMW 当初の依頼のワイバーン退治は達成できたが……より厄介なモノを呼び寄せたのかもしれない
	PRINTFORMW 後味の悪さを感じながら、%QST_LEADER_NAME()%は報告のために牧羊の村に戻った……
	PRINTW
	CALL FL_STOPBGM()
	CALL FL_PLAYSE("Fanfare_0.ogg")
	PRINTFORMW ――『飛竜の群れ』を討伐した！
	PRINTL
	;戦闘で増減したステータスをキャラに反映
	CALL CC_APPLY_CHARA_HP()
	CALL QST_CHANGE_ITEM("MONEY,20000,名誉点,300,")
	;戦利品獲得処理
	CALL CC_GET_LOOT_FROM("DEFENDER")
	QST_SV_QUEST_FLAG:QST_有効クエストID = 2
	CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)
	;クエスト終了時の決まり文句
	TFLAG:クエスト_完了フラグ = 1
	CALL TRAIN_EXIT()
ELSE
	;敗北イベント
	CALL CC_APPLY_CHARA_HP()
	PRINTL
	PRINTFORMW 呼吸のたびに四肢に充足する莫大な魔力
	PRINTFORMW 魔力を流すたびに硬質化し、こちらの攻撃を弾く屈強な鱗の鎧
	PRINTL
	PRINTFORMW 腕の一振りで数人を撃ち払い、足の一踏みが大地に破砕の亀裂を刻む
	PRINTFORMW まさに暴虐の化身と言うに相応しい暴れっぷりに、%QST_LEADER_NAME()%の消耗は色濃い
	PRINTFORMW このままでは勝てない……！
	PRINTL
	PRINTFORMW ギリギリの情勢下で、%QST_LEADER_NAME()%は退却を決意した
	PRINTL
	PRINTFORML 狭い洞窟内での撤退戦となったが……
	PRINTFORMW 敵の大ぶりな攻撃と、狭所でのぎこちない動きが幸いし、なんとか逃げることができた
	PRINTL
	PRINTFORMW %QST_LEADER_NAME()%は這う這うの体で、森へと落ち延びるのだった
	PRINTL
	CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)
	;ゲームオーバー共通イベント
	CALL QST_QUEST_23_DEFEAT_EVENT
ENDIF

;-------------------------------------------------
;以下イベントリストで管理する必要のない処理,イベント固有の関数など
;-------------------------------------------------
;汎用敗北イベント
@QST_QUEST_23_DEFEAT_EVENT
PRINTL
PRINTFORMW ……………………
PRINTFORMW ………………
PRINTFORMW …………
PRINTL

PRINTFORMW %QST_LEADER_NAME()%はボロボロの身を引きずって、なんとか牧羊の村まで帰還した……
PRINTFORML
PRINTFORMW ……今回の飛竜退治は失敗に終わってしまった。帰還して一度態勢を立て直すことにしよう
PRINTL
CALL PRINT_NOTICE("治療費や宿泊費として10,000ディナリを失った……")
MONEY -= 10000
QST_SV_QUEST_FLAG:QST_有効クエストID = 1
TFLAG:クエスト_完了フラグ = 1
CALL TRAIN_EXIT()

;-------------------------------------------------
;@QST_QUEST_23_ENEMY_DATA
; クエスト固有エネミーの読み込み処理
; クエストで共通的に使い回すエネミーの戦闘データは@CC_ENEMY_DATAに記述されているが
; クエスト側で後付で追加することもできる。(@QST_INIT_QUEST_XXなどで明示的に読込み処理を呼び出すこと)
; 特定のクエストでしか使わないエネミーのデータ（ボスなど）はクエスト側に記述するとよいだろう
; パラメータの詳細は @CC_ENEMY_DATA 参照
; ！！！ENEMY_DEF_NAMEはユニークである必要があるためクエスト側でエネミーを追加する場合はENEMY_DEF_NAMEに'QUEST_XX_'を接頭すること！！！
;-------------------------------------------------
@QST_QUEST_23_ENEMY_DATA
{
	DT_ROW_ADD "ENEMY_DATA"
		,"ENEMY_DEF_NAME", "QUEST_23_謎の少女"
		,"MAX_HP", 25000
		,"STR", 220
		,"MAG", 125
		,"ATTACK", 220
		,"DEFENCE", 55
		,"ARMOR", 500
		,"AGILITY", 200
		,"HATE_RATE", 100
		,"TARGETING", WPN_脅威選定_魔術師
		,"ENEMY_AI", "[SKILL_USE_RATE:70]"
		,"REGIST_POISON", 50
		,"REGIST_STUN", 90
		,"REGIST_BURN", 0
		,"REGIST_CHARM", 70
		,"REGIST_PANIC", 70
		,"REGIST_BLIND", 70
		,"REGIST_DEATH", 100
		,"ENEMY_PASSIVE_SKILL", "[ENEMY_SKILL_DRAGONHEART:2][ENEMY_SKILL_DRAGONSCALE:1]"
		,"ENEMY_ACTIVE_SKILL", "[ENEMY_SKILL_THUNDER_CROW:2][ACT_REACT:3]"
		,"DROP_ITEM_LIST", @"{ITEM_技巧の宝玉},3,1000,{ITEM_精霊石},2,1000,"
		,"EXP", 50
}
{
	DT_ROW_ADD "ENEMY_DATA"
		,"ENEMY_DEF_NAME", "QUEST_23_ダイアウルフの幻影"
		,"MAX_HP", 6500
		,"STR", 60
		,"MAG", 30
		,"ATTACK", 60
		,"DEFENCE", 40
		,"ARMOR", 400
		,"AGILITY", 150
		,"HATE_RATE", 100
		,"TARGETING", WPN_脅威選定_無作為
		,"ENEMY_AI", "[SKILL_USE_RATE:50]"
		,"ENEMY_ACTIVE_SKILL", "[SMITE:2][ACT_FRONT_DEFENCE:3]"
		,"DROP_ITEM_LIST", @"{ITEM_銀インゴット},1,1000,"
		,"EXP", 15
}
{
	DT_ROW_ADD "ENEMY_DATA"
		,"ENEMY_DEF_NAME", "QUEST_23_ウォッチャー"
		,"MAX_HP", 3000
		,"STR", 55
		,"MAG", 10
		,"ATTACK", 55
		,"DEFENCE", 20
		,"ARMOR", 200
		,"AGILITY", 120
		,"HATE_RATE", 100
		,"TARGETING", WPN_脅威選定_低脅威
		,"REGIST_POISON", 0
		,"REGIST_STUN", 0
		,"REGIST_BURN", 0
		,"REGIST_CHARM", 0
		,"REGIST_PANIC", 0
		,"REGIST_BLIND", 100
		,"REGIST_DEATH", 100
		,"ENEMY_AI", "[SKILL_USE_RATE:25]"
		,"ENEMY_ACTIVE_SKILL", "[ACT_TORN_CLOTH:1][ACT_STUN_ENEMY:1]"
		,"DROP_ITEM_LIST", @"{ITEM_魔獣の肉片},1-2,350,{ITEM_マナ結晶},1-3,400,"
		,"EXP", 3
}


code:
;-------------------------------------------------
;クエスト定義ファイル
;QUEST_24
;-------------------------------------------------
@QST_TITLE_QUEST_24
RESULTS '= "冒険者シルフィーユ"

;ギルドメニューでクエストをフィルターする際のタグリスト。末尾の','を忘れないように
@QST_QUEST_TAG_24
RESULTS '= "単発,イベント,"

;-------------------------------------------------
;クエストの実行可能条件を定義
;-------------------------------------------------
@QST_AVAILABLE_QUEST_24(QUEST_ID)
#DIM QUEST_ID
;シルフィーユが仲間なら実行しない
SIF CFLAG:NAME_TO_CHARA("UC_シルフィーユ"):所属 == 所属_プレイヤー
    RETURN 0
;前提としてQUEST_25_開拓村の魔物退治の攻略が必要
SIF QST_SV_QUEST_FLAG:25 != 2
    RETURN 0
RETURN 1

;-------------------------------------------------
;クエストのタイプ
;マップ探索を伴う場合 "MAP"
;文章のみで完結する場合 "DIALOGUE"
;-------------------------------------------------
@QST_TYPE_QUEST_24
RESULTS '= "DIALOGUE"

;クエストメニュー右上段の概要欄。10行まで
@QST_SUMALLY1_QUEST_24
RESULTS:0 '= "開拓村で共に魔物退治をした冒険者の剣士、シルフィーユが近くに滞在しているそうだ"
RESULTS:1 '= "彼女を仲間にしたいのであれば、勧誘してみるのもいいだろう"

;クエストメニュー右下段の備考欄。難易度や条件、報酬等を記述。7行まで
@QST_SUMALLY2_QUEST_24
RESULTS:0 '= @"難易度：-"
RESULTS:1 '= "雇用費：-"
RESULTS:3 '= @"@C:{カラー_選択不可}@「立ち振る舞いからして、結構いいところの嬢ちゃんっぽいな」@/C@"
RESULTS:4 '= @"@C:{カラー_選択不可}@「なんでわざわざこんな辺境で冒険者しているのかは知らんが」@/C@"
RESULTS:5 '= @"@C:{カラー_選択不可}@「裏があるような嬢ちゃんじゃねえだろう、良くも悪くもな」@/C@"

;-------------------------------------------------
;QST_INTRO_QUEST_n はQUEST開始前に呼ばれる
;クエストの導入文などを記述
;"DIALOGUE"型のクエストは当該関数のCALLのみ行う
;-------------------------------------------------
@QST_INTRO_QUEST_24
#DIM WEAPON_LIST_IDX
#DIM CHARA_ID

PRINTFORMW 「%ANAME(MASTER)%さんっ、ご無沙汰してますっ！　元気そうでなによりです！」
PRINTFORMW シルフィーユ――フィーは元気よく挨拶してくれた
PRINTFORMW しばしの間、彼女との会話を楽しむ
PRINTL
PRINTFORMW どうやら、彼女は%ANAME(MASTER)%の領地の噂を聞きつけて、見識を広める一環として近くまで来たようだ
PRINTFORMW この近くの町でしばらく冒険者としての活動を予定しているようだが、それを強く希望しているようでもない
PRINTL
PRINTFORMW 前回の依頼を共にして、彼女の冒険者としての力量や人格は保証されている
PRINTFORMW 仲間に誘って見るのも良さそうだ

PRINTL
CALL ASK_YN("仲間に誘う","やめておく")
IF !RESULT
    PRINTFORML %ANAME(MASTER)%はフィーを仲間に誘うことにした
    PRINTFORMW よければ、うちの拠点に来てくれないだろうか。一緒に冒険がしたい
    PRINTL
    PRINTFORMW 「わ、私を仲間にですかっ！？」
    PRINTL
    PRINTFORML 彼女は最初は驚いたようだが、少し考えた後、笑顔で頷いた
    PRINTFORMW 「誘ってくださって、ありがとうございます。こちらこそ、ぜひ仲間にさせてくださいっ！」
    PRINTL
    PRINTFORML こちらこそよろしくと返す。
    PRINTFORMW それと、これからは仲間なのだから敬語はいらないと告げる
    PRINTL
    PRINTFORML 「そ、そうですか？　分かりましたっ！」
    PRINTFORMW 「……じゃなかった。うん、分かったよ」
    PRINTL
    PRINTFORMW 「これからよろしくね、%ANAME(MASTER)%！」
    PRINTL
    CALL PRINT_NOTICE("シルフィーユが仲間になった")
    ;加入処理
    CHARA_ID = NAME_TO_CHARA("UC_シルフィーユ")
    CFLAG:CHARA_ID:所属 = 所属_プレイヤー
    ;初期装備の作成と装備
    CALL QST_GET_WEAPON(2, "シルフィーユの鉄の長剣", WPN_材質_鉄,,,,0)
    WEAPON_LIST_IDX = RESULT
    CALL EQUIP_WEAPON(CHARA_ID, WEAPON_LIST_IDX)
    QST_SV_QUEST_FLAG:QST_有効クエストID = 1
ELSE
    PRINTFORMW フィーにはフィーの都合もあるだろう。無理に誘う必要もない
    PRINTFORMW %ANAME(MASTER)%はそう判断した
    PRINTL
    PRINTFORMW 「良ければ、また会いましょう！　%ANAME(MASTER)%さん！」
    PRINTFORMW %ANAME(MASTER)%はフィーに頷き、帰路についた
ENDIF

TFLAG:クエスト_完了フラグ = 1
CALL TRAIN_EXIT()

path: ERB/QUEST/クエストデータ/QUEST_24_冒険者シルフィーユ.ERB